add trrexfpf3gpphsi3d.cw

add trrexfpf3gpphsi3d.cw

;15N constant-time TROSY CPMG relaxation dispersion with temperature compensation, pseudo-3D (recommended version)
;Only for use with Topspin3
;Based on trrexfpf3gpphsi3d.jk
;
;P. Vallurupalli, D.F. Hansen, E. Stollar, E. Meirovitch & L.E. Kay,
;PNAS 104, 18473-18477 (2007)
;
;John Kirkpatrick, UCL, Feb 2014
;
;Temperature compensation element after recycle delay
;See end of sequence for available options
;Added element to check that d21 and field strengths in vdlist file are all compatible, 12/5/14
;Removed assignment of vd entries to loop counters (to allow non-integer field strengths), 14/5/14
;
;-----------------------------------------------------------------------------------
;
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple2>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p22=p21*2"

# ifdef CAL_N
    define pulse pNcal
    "pNcal=p25/2"
# endif /*CAL_N*/

"d11=30m"
"d12=20u"
"d13=4u"

"d26=1s/(cnst4*4)"

# ifndef ONE_D
"in10=inf2/4"
"d10=in10/2-5u"
# endif /*ONE_D*/

define delay VDCOMP
define loopcounter COUNTER1

;----parameters needed to check that CPMG counter is always naturally integer for given d21/vdlist

define loopcounter VDMIN
define loopcounter D21_USEC

"D21_USEC=d21*1000000 + 0.5"	; length of d21 in microseconds (integer)
"VDMIN=500000000/D21_USEC"	; minimum allowed CPMG field-strength x 1000 (integer)

;----

"DELTA1=d25-p16-d16-larger(p1,p21)-4u"
"DELTA2=d26-p16-d16+2u"
"DELTA3=d25-p19-d16-larger(p1,p21)-d12-p11-p1"
"DELTA4=d25-p19-d16-larger(p1,p21)-d12-p11-4u"
"DELTA5=d25-p19-d16-larger(p1,p21)-d12-p11-p21-4u"

"plw23=plw3*(p22/p25)*(p22/p25)*cnst63"		; power for CPMG train

# ifndef ANTI_TROSY
    "TAU2=4*p1+4u"
# endif /*ANTI_TROSY*/

"l0=1"

"spoff1=0"
"spoff11=0"

# ifdef LABEL_CN
    "spoff13=0"
# endif /*LABEL_CN*/

aqseq 312

1 ze
  d11

  if "d1 < 2.0s"
    {
    d11
    print "error: recycle delay too short, d1 must be >= 2.0 s, aborting..."
    goto stop
    }

  if "d21 > 25.1m"
    {
    d11
    print "error: CPMG train too long, d21 must be =< 25 ms, aborting..."
    goto stop
    }

  if "p25 < 80u"
    {
    d11
    print "error: CPMG 180deg pulse too short, p25 must be >= 80 us, aborting..."
    goto stop
    }

# ifdef INTERLEAVE
    d11 st0
# else
    define loopcounter PHASE
    define loopcounter VD_CNT
    define loopcounter T1_CNT
    d12
    "VD_CNT=td1"
    if "td2 == 1"
      {
      d12
      "PHASE=1; T1_CNT=1"
      }
    else
      {
      d12
      "PHASE=2; T1_CNT=td2/2"
      }
# endif /*INTERLEAVE*/

2 20m
3 d11
4 12m
5 d13

# if defined(LABEL_CN) && !defined (ONE_D)
    if "4*d10-24u > p8"
      {
      d12
      "d60=d10-0.25*p8"
      }
    d12 pl0:f2
# endif /*LABEL_CN, ONE_D*/

  if "vd > 1500"
    {
    d11
    print "error: CPMG frequency too high, must be < 1500, aborting..."
    goto stop
    }

  if "(1000*vd)%VDMIN != 0"
    {
    d11
    print "error: One or more field strengths in vdlist file does not give integer loop counter \
           for CPMG train - check vdlist and/or d21.  Aborting..."
    goto stop
    }

  if "vd == 0"
    {
    20u
    "TAU=1s"
    20u
    "COUNTER=0"
    }
  else
    {
    20u
    "TAU=(1 / (vd*4) ) - p25/2000000"
    20u
    "COUNTER=d21*vd*2 + 0.1"
    }

  d1

# ifndef NOCOMP

  20u
  "VDCOMP=cnst62-vd"

  if "cnst62 > 1500"
    {
    d11
    print "error: maximum CPMG frequency as specified by cnst62 is too high, must be < 1500, aborting..."
    goto stop
    }

  if "VDCOMP < 0"
    {
    d11
    print "error: cnst62 is less than maximum CPMG frequency in vdlist, aborting..."
    goto stop
    }

  if "VDCOMP == 0"
    {
    20u
    "TAU1=1s"
    20u
    "COUNTER1=0"
    }
  else
    {
    20u
    "TAU1=(1 / (VDCOMP*4) ) - p25/2000000"
    20u
    "COUNTER1=d21*VDCOMP*2 + 0.1"
    }

  if "VDCOMP == 0"
    {
    d12
    d21*2
    }
  else
    {
    if "vd == 0"
      {
      d21*2
      }
    d12 pl23:f3
6   TAU1
    (p25 ph21):f3
    TAU1
    TAU1
    (p25 ph21):f3
    TAU1
    lo to 6 times COUNTER1
    }

# endif /*NOCOMP*/

  50u UNBLKGRAD
  d12 pl3:f3
  p16:gp0*0.7
  d16
  (p21 ph20):f3
  4u
  p16:gp0
  d16*2

  d12 pl0:f1
  (p11:sp1 ph13:r):f1		; flipdown(+y): +z -> +x
  d12 pl1:f1

# ifdef ANTI_TROSY
    (p1 ph10)
# else
    (p1 ph12)
# endif /*ANTI_TROSY*/

  4u
  p16:gp1
  d16
  DELTA1
  (center (p2 ph10) (p22 ph20):f3 )
  DELTA1
  4u
  p16:gp1
  d16
  (p1 ph11)

  4u
  p16:gp2
  d16*2
  d12 pl23:f3

  d8			; tau_eq

  (p25*0.5 ph1):f3

  if "vd > 0"
    {
7   TAU
    (p25 ph21):f3
    TAU
    lo to 7 times COUNTER
    }

  (p25*0.5 ph2):f3

  DELTA2
  p16:gp3
  d16 pl3:f3

  (p1 ph11 2u p2 ph10 2u p1 ph11):f1
  2u

# ifdef ANTI_TROSY
    (p21 ph20 2u p22 ph21 2u p21 ph20):f3	; 180(y)
# else
    (p21 ph21 2u p22 ph20 2u p21 ph21):f3	; 180(x)
# endif /*ANTI_TROSY*/

  DELTA2
  p16:gp3
  d16 pl23:f3

  (p1 ph11 2u p2 ph10 2u p1 ph11):f1
  2u
  (p25*0.5 ph3):f3

  if "vd > 0"
    {
8   TAU
    (p25 ph20):f3
    TAU
    lo to 8 times COUNTER
    }

  (p25*0.5 ph21):f3

# ifdef CAL_N
    4u
    p16:gp8*0.7
    d16
    (pNcal ph20):f3
    4u
    p16:gp8
    d16
# endif /*CAL_N*/

  d8			; tau_eq

# ifdef ANTI_TROSY
    (p1 ph11 2u p2 ph10 2u p1 ph11):f1
# else
    TAU2
# endif /*ANTI_TROSY*/

  4u
  p16:gp4
  d16*2
  d12 pl3:f3

  if "l0 %2 == 1"
    {
    (p21 ph4):f3
    }
  else
    {
    (p21 ph14):f3
    }

# ifndef ONE_D

# ifdef LABEL_CN

    if "4*d10-24u < p8"
      {
      2u
      d10 gron5
      d10 gron5*-1
      10u groff
      d10 gron5
      d10 gron5*-1
      8u groff
      }
      else
      {
      2u
      d60 gron5
      d60 gron5*-1
      8u groff
      (p8:sp13 ph10):f2
      2u
      d60 gron5
      d60 gron5*-1
      8u groff
      }

# else /*LABEL_CN*/

    2u
    d10 gron5
    d10 gron5*-1
    10u groff
    d10 gron5
    d10 gron5*-1
    8u groff

# endif /*LABEL_CN*/

# endif /*ONE_D*/

  (p1 ph5)

  d12 pl0:f1

# ifdef ANTI_TROSY
    (p11:sp11 ph5:r):f1		; flipback(+y): -x -> +z
# else
    (p11:sp11 ph6:r):f1         ; flipback(-y): +x -> +z
# endif /*ANTI_TROSY*/

  DELTA3
  p19:gp6
  d16 pl1:f1

  (center (p2 ph10) (p22 ph20):f3 )

  4u
  p19:gp6
  d16 pl0:f1
  DELTA4

  (p11:sp11 ph10:r):f1		; flipback(+x): -z -> +y
  d12 pl1:f1

  (center (p1 ph10) (p21 ph7):f3 )

  4u
  p19:gp7
  d16 pl0:f1
  DELTA4

  (p11:sp1 ph12:r):f1		; flipdown(-x): +z -> +y
  d12 pl1:f1

  (center (p2 ph10) (p22 ph20):f3 )

  d12 pl0:f1
  (p11:sp11 ph22:r):f1		; flipback(-x): -y -> +z

  DELTA5
  p19:gp7
  d16 pl1:f1

  4u BLKGRAD
  (p21 ph20):f3

# ifdef INTERLEAVE

  goscnp ph31
  3m st ivd
  lo to 2 times nbl

  20m ipp1 ipp2 ipp3 ipp4 ipp14 ipp31
  lo to 3 times ns

  d11 mc #0 to 3
     F1QF()
     F2EA(calph(ph5, +180) & calph(ph6, +180) & calph(ph7, +180) & calph(ph31, +180) & calclc(l0, 1) & exec(rppall), caldel(d10, +in10) & calph(ph4, +180) & calph(ph14, +180) & calph(ph31, +180))

# else /*INTERLEAVE*/

  gosc ph31
  3m
  lo to 2 times ns
  10m wr #0 if #0 zd 10m ivd
  lo to 3 times VD_CNT
  6m ip5*2 6m ip6*2 6m ip7*2 6m ip31*2 6m iu0
  lo to 4 times PHASE
  3m id10 3m ip4*2 3m ip14*2 3m ip31*2
  lo to 5 times T1_CNT

;;--------------------------- this construct is not working in Topspin 3.2.5  ---------------------------
;  d11 mc #0 to 3
;     F1QF(ivd)
;     F2EA(calph(ph5, +180) & calph(ph6, +180) & calph(ph7, +180) & calph(ph31, +180) & calclc(l0, 1), caldel(d10, +in10) & calph(ph4, +180) & calph(ph14, +180) & calph(ph31, +180))
;;-------------------------------------------------------------------------------------------------------

# endif /*INTERLEAVE*/

stop, exit

; Proton/carbon phases

ph5= 1
ph6= 3
ph10=0
ph11=3
ph12=2
ph22=2
ph13=1

; Nitrogen phases

ph1= 0 2
ph2= 3 3 1 1
ph3= 0 0 2 2
ph4= 3 3 1 1 2 2 0 0
ph14=3 3 1 1 0 0 2 2
ph7= 1
ph20=0
ph21=3

; Receiver phase

ph31=3 1 1 3 0 2 2 0

;;-------------Varian phases----------------
;
;; Proton phases
;
;ph5= 3
;ph6= 1
;ph10=0
;ph11=1
;ph12=2
;ph13=3
;
;; Nitrogen phases
;
;ph1= 0 2
;ph2= 1 1 3 3
;ph3= 0 0 2 2
;ph4= 1 1 3 3 2 2 0 0
;ph14=1 1 3 3 0 0 2 2
;ph7= 3
;ph20=0
;ph21=1
;
;; Receiver phase
;
;ph31=1 3 3 1 0 2 2 0
;
;;------------------------------------------

;pl0 : 120dB
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl23: f3 channel - power level for CPMG
;sp1:  f1 channel - shaped pulse  90 degree [flipdown]
;sp11: f1 channel - shaped pulse  90 degree [flipback]
;sp13: f2 channel - shaped pulse 180 degree (adiabatic)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p11: f1 channel -  90 degree shaped pulse
;p16: homospoil/gradient pulse                       [1 msec]
;p19: gradient pulse 2                               [500 usec]
;p21: f3 channel -  90 degree high power pulse
;pNcal: f3 channel - calibration 90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p25: f3 channel - 180 degree pulse at pl23          [90 usec]
;d1 : relaxation delay; 1-5 * T1
;d8 : equilibration delay
;d10: incremented delay                              [3 usec]
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21: length of single CPMG train
;     total constant-time relaxation period is 2*d21
;d25: 1/(4J)YH for YH
;d26: 1/(4J(YH))
;d60: calculated internally
;vd : variable field strength, taken from vd-list
;cnst4: = J(YH)
;cnst62: maximum CPMG frequency in vd-list
;cnst63: multiplicative correction to power for 15N CPMG (calibrate)
;vd: taken from vdlist (CPMG field strength)
;VDCOMP: calculated internally (CPMG field strength for compensation element)
;inf2: 1/SW(X) = 4 * DW(X)
;in10: 1/(4 * SW(X)) = DW(X)
;nd10: 4
;NS: 8 * n
;DS: >= 32
;td1: number of delays in vd-list
;td2: number of experiments in F2
;NBL: = td1
;FnMODE: QF in F1
;FnMODE: echo-antiecho in F2

;for z-only gradients:
;gpz0: -37%
;gpz1: 11%
;gpz2: 29%
;gpz3: 7%
;gpz4: 19%
;gpz5: 1-2%
;gpz6: 23%
;gpz7: 43%
;gpz8: 71%

;use gradient files:
;gpnam0: SINE.100
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam6: SINE.100
;gpnam7: SINE.100
;gpnam8: SINE.100

;note: the values in the vd-list are interpreted as field-strength in Hz

                                          ;preprocessor-flags-start
;INTERLEAVE: for looping over vd-list before ns
;LABEL_CN: for 13C decoupling during indirect 15N evolution period
;CAL_N: for calibration of 180deg pulse for 15N CPMG
;ONE_D: for running 1D or pseudo-2D version (no 15N chemical shift evolution)
;NOCOMP: for running without compensation elements
;ANTI_TROSY: for measuring anti-TROSY line during CPMG
                                          ;preprocessor-flags-end

error: pathspec 'trrexfpf3gpphsi3d.cw' did not match any file(s) known to git


adding (and tested) cosyph.cw

adding (and tested) cosyph.cw

;cosyph
;avance-version (12/01/11)
;2D homonuclear shift correlation
;phase sensitive
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>

"in0=inf1"

"d0=in0/2-p1*4/3.1416"
"acqt0=-p1*2/3.1416"

1 ze
2 d1
3 p1 ph1
  d0
  p1 ph2
  go=2 ph31
  d1 mc #0 to 2 F1PH(calph(ph1, +90), caldel(d0, +in0))
exit

ph1=0 2 2 0 1 3 3 1
ph2=0 2 0 2 1 3 1 3
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;p0 : f1 channel -  20 to 90 degree high power pulse
;p1 : f1 channel -  90 degree high power pulse
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;inf1: 1/SW = 2 * DW
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

;$Id: cosyph,v 1.9 2012/01/31 17:49:22 ber Exp $

error: pathspec 'cosyph.cw' did not match any file(s) known to git


add noesyhsqcgpsm3d.2.cw

add noesyhsqcgpsm3d.2.cw

;noesyhsqcgpsm3d.2
;avance-version (12/01/11)
;NOESY-HSQC
;3D sequence with
;   homonuclear correlation via dipolar coupling
;   dipolar coupling may be due to noe or chemical exchange.
;   H-1/X correlation via double inept transfer
;   simultaneous evolution of C-13 and N-15 chemical shift in t2
;phase sensitive (t1)
;phase sensitive (t2)
;with decoupling during acquisition
;using shaped pulses for inversion on f2 - channel
;(use parameterset NOESYHSQCGPSM3D.2)
;
;S.M. Pascal, D.R. Muhandiram, T. Yamazaki, J.D. Forman-Kay & L.E. Kay,
;   J. Magn. Reson. B103, 197 - 201 (1994)
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"p22=p21*2"
"d4=1s/(cnst2*4)"
"d11=30m"
"d12=20u"
"d26=1s/(cnst4*4)"
"d29=3u"

"p16=1m"
"p19=3m"
"p29=4m"

"d0=3u"
"d10=3u"

"in0=inf1/2"
"in10=inf2/2"

"in29=in30/2-in10"

"DELTA=d1-d11*4-6m"
"DELTA1=d4-p16-d16-p8/2"
"DELTA2=d26-d4"
"DELTA3=d4-p16-d16-p8/2-p3*2-p21*2-d12-10u"
"DELTA4=d0*2+larger(p8,p22)"
"DELTA5=p2*2+p3*2+p4+d10*4+d29*2"
"DELTA6=d10*2+p2"

"TAU=d8-p19*2-d16*2-p3"
"TAU2=(p8-p2)/2"
"TAU3=(p8-p22)/2"

"spoff13=0"

aqseq 321

1 ze
  d11 pl12:f2 pl16:f3
2 d11 do:f2 do:f3
3 d12 pl10:f1
  p17 ph1
  p17*2 ph2

  d1 pl1:f1 pl0:f2 pl3:f3
  50u UNBLKGRAD
  p16:gp7
  d16

  (p1 ph3)
  DELTA4
  (p2 ph8)
  d0
  (center (p8:sp13 ph1):f2 (p22 ph1):f3 )
  d0
  (p1 ph1)

  TAU
  p19:gp1
  d16 pl2:f2
  (p3 ph1):f2
  4u
  p16:gp2
  d16

  (p1 ph1)
  4u
  p16:gp3
  d16 pl0:f2
  DELTA1
  (lalign
   (DELTA2 TAU2 p2 ph1 TAU2)
   (p8:sp13 ph1):f2
   (DELTA2 TAU3 p22 ph1 TAU3):f3
  )
  DELTA2
  DELTA1
  4u
  p16:gp3
  d16
  (p1 ph2)

  4u
  p29:gp4
  d16 pl2:f2

  (p21 ph5):f3
  d29
  (p3 ph4):f2
  d10
  (p2 ph7)
  d10
  (p4 ph9):f2
  DELTA6
  (p3 ph1):f2
  d29
  (p22 ph9):f3
  DELTA5
  (p21 ph1):f3

  4u
  p16:gp5
  d16

  (p1 ph6)
  4u
  p16:gp6
  d16 pl0:f2
  DELTA1
  (lalign
   (DELTA2 TAU2 p2 ph1 TAU2):f1
   (p8:sp13 ph1):f2
   (DELTA2 TAU3 p22 ph1 TAU3):f3
  )
  DELTA2
  DELTA3
  4u
  p16:gp6
  d16 pl2:f2
  (p3 ph1 3u p3 ph7):f2
  (p21 ph1 3u p21 ph7):f3
  d12 pl12:f2 pl16:f3
  4u BLKGRAD
  (p1 ph6)

  go=2 ph31 cpd2:f2 cpd3:f3
  d11 do:f2 do:f3 mc #0 to 2
     F1PH(calph(ph3, +90) & calph(ph8, +90), caldel(d0, +in0))
     F2PH(calph(ph4, +90) & calph(ph5, +90), caldel(d10, +in10) & caldel(d29, +in29))
exit

ph1=0
ph2=1
;ph3=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph3=0 0 0 0 2 2 2 2
ph4=0 2
ph5=2 0
;ph6=0 0 1 1 2 2 3 3
ph6=0 0 1 1 0 0 1 1 2 2 3 3 2 2 3 3
;ph7=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph7=0 0 0 0 2 2 2 2
;ph8=1 1 1 1 1 1 1 1 3 3 3 3 3 3 3 3
ph8=1 1 1 1 3 3 3 3
ph9=0 0 1 1
ph31=0 2 1 3 2 0 3 1 2 0 3 1 0 2 1 3

;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl10: f1 channel - power level for TOCSY-spinlock (trim pulse)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl16: f3 channel - power level for CPD/BB decoupling
;sp13: f2 channel - shaped pulse 180 degree (adiabatic)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p16: homospoil/gradient pulse                       [1 msec]
;p17: f1 channel - trim pulse                        [4 msec]
;p19: gradient pulse 2                               [3 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p29: gradient pulse 3                               [4 msec]
;d0 : incremented delay (F1 in 3D)                   [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4J(CH))
;d8 : mixing time
;d10: incremented delay (F2 in 3D)                   [3 usec]
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d26: 1/(4J(NH))
;d29: decremented delay (F2 in 3D)                   [3 usec]
;cnst2: = J(CH)
;cnst4: = J(NH)
;inf1: 1/SW(H) = 2 * DW(H)
;inf2: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(H)) =  DW(H)
;nd0: 2
;in10: 1/(2 * SW(C)) = DW(C)
;nd10: 2
;in29: in30/2 - in10
;in30: 1/SW(N) = 2 * DW(N)
;ns: 8 * n
;ds: >= 16
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or TPPI) in F1
;FnMODE: States-TPPI (or TPPI) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;use gradient ratio:    gp 1 : gp 2 : gp 3 : gp 4 : gp 5 : gp 6 : gp 7
;                         30 :   40 :   16 :   60 :  -36 :   16 :   50

;for z-only gradients:
;gpz1: 30%
;gpz2: 40%
;gpz3: 16%
;gpz4: 60%
;gpz5: -36%
;gpz6: 16%
;gpz7: 50%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
;gpnam6: SMSQ10.100
;gpnam7: SMSQ10.100

;$Id: noesyhsqcgpsm3d.2,v 1.9 2012/01/31 17:49:28 ber Exp $

error: pathspec 'noesyhsqcgpsm3d.2.cw' did not match any file(s) known to git


adding project_ste.cw

adding project_ste.cw

; project_ste.cw
; PROJECT-STE T2-D correlation sequence
;
; based on J.A. Aguilar, M. Nilsson, G. Bodenhausen and G. Morris
; Chem. Commun. (2012) 48 811-813
;
; run as pseudo-3D
; with presaturation
;
; Chris Waudby
; July 2018

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

prosol relations=<triple>

;define list<gradient> diff=<Difframp>
define list<gradient> diff={0.089 0.096 0.121 0.164 0.214 0.271 0.332 0.396 0.463 0.532 0.604 0.675 0.750 0.827 0.904 0.982}
;define list<loopcounter> ncyc = {0 1 2 4 6 8 10 13 16 19 22 26 29 33 37 41}
define list<loopcounter> ncyc = {0 41 1 37 2 33 4 29 6 26 8 22 10 19 13 16}

"p2=p1*2"
"d2=250u"
"d3=4*d2"  ; T2 time per loop
"d12=20u"
"DELTA1=d2-p1"
"DELTA2=d2-1.5*p1"
"TAU=d20-p30*2-p19-d16*3-p11-p1*4-8u"
"acqt0=-0.6366*p1"

; acquisition order = D then T2
aqseq 321

1 ze
2 30m pl9:f1
  d1 cw:f1 ph29
  4u do:f1
  d12 pl1:f1 UNBLKGRAD

  ; start PROJECT echos
  p1 ph11

3 DELTA1
  p2 ph2
  DELTA2
  p1 ph12
  DELTA2
  p2 ph2
  DELTA1
  lo to 3 times ncyc

  ; STE encoding
  p30:gp6*diff
  d16
  p2 ph14
  p30:gp6*-1*diff
  d16
  p1 ph15

  ; diffusion delay
  4u
  p19:gp7
  d16 pl9:f1
  TAU cw:f1 ph1
  4u do:f1

  ; water flip-down and STE decoding
  (p11:sp1 ph23:r)
  4u pl1:f1
  p1 ph13
  p30:gp6*diff
  d16
  p2 ph1
  p30:gp6*-1*diff
  d16
  4u BLKGRAD

  ; acquisition
  go=2 ph31
  30m mc #0 to 2
      F2QF(igrad diff)
      F1QF(ncyc.inc)
exit

ph1 =0
ph2 =1
ph11=0 2
ph12=1 1 3 3
ph13=0 0 0 0  1 1 1 1  2 2 2 2  3 3 3 3
ph23=2 2 2 2  3 3 3 3  0 0 0 0  1 1 1 1
ph14={0}*16 {2}*16
ph15={0}*32 {2}*32
ph29=0
ph31={0 2 0 2  1 3 1 3  2 0 2 0  3 1 3 1}*2
     {2 0 2 0  3 1 3 1  0 2 0 2  1 3 1 3}*2

;pl1: f1 channel - power level for pulse (default)
;pl9: f1 channel - power level for presaturation
;p1 : f1 channel - 90 degree excitation pulse
;p2 : f1 channel - 180 degree refocusing pulse
;p19: gradient pulse 2 (spoil gradient)
;p30: gradient pulse (little DELTA * 0.5) [2.75ms]
;d1 : relaxation delay; 1-5 * T1
;d2 : tau (T2 time = 4*d2*vc)
;d3 : total T2 time per loop
;d12: delay for power switching [20 usec]
;d16: delay for gradient recovery
;d20: diffusion time (big DELTA)
;ds : 16
;ns : 32*n

;for z-only gradients:
;gpz6: 100%
;gpz7: -17.13% (spoil)

;use gradient files:
;gpnam6: SMSQ10.100
;gpnam7: SINE.100

;use AU-program dosy to calculate gradient-file Difframp

error: pathspec 'project_ste.cw' did not match any file(s) known to git


add stebpgp_f2dec.cw, 1H STE with 13C decoupling

;with 13C CPD during acquisition
;With added presaturation (during recycle delay only)
;Zero first-order phase correction
;
;stebpgp1s
;avance-version (07/05/08)
;2D sequence for diffusion measurement using stimulated echo
;using bipolar gradient pulses for diffusion
;using 1 spoil gradient
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple_d>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define list<gradient> diff=<Difframp>

"p2=p1*2"

"d11=30m"
"d12=20u"
"d13=4u"

"TAU=8u+de-0.6366*p1"
"DELTA1=d20-p1*2-p2-p30*2-d16*3-p19-TAU"

"acqt0=de"

1 ze
  4u pl12:f2
  d12 pl1:f1
2 d1
  50u UNBLKGRAD
  p1 ph1
  p30:gp6*diff
  d16
  p2 ph2
  p30:gp6*-1*diff
  d16
  p1 ph3
  p19:gp7
  d16
  DELTA1
  p1 ph4
  TAU
  p30:gp6*diff
  d16
  p2 ph2
  p30:gp6*-1*diff
  d16 pl12:f2
  8u BLKGRAD
  go=2 ph31 cpd2:f2
  d11 do:f2 mc #0 to 2 F1QF(igrad diff)
exit

ph1= 0
ph2= 0 0 0 0  2 2 2 2
ph3= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph4= 0 1 2 3
ph29=0
ph31=0 3 2 1  0 3 2 1  2 1 0 3  2 1 0 3

;pl1: f1 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p19: gradient pulse 2 (spoil gradient)
;p30: gradient pulse (little DELTA * 0.5)
;d1 : relaxation delay; 1-5 * T1
;d16: delay for gradient recovery
;d20: diffusion time (big DELTA)
;NS : 8 * n
;DS : 4 * m
;td1: number of experiments
;FnMODE: QF
;        use xf2 and DOSY processing

;use gradient ratio:    gp 6 : gp 7
;                       100  : -17.13

;for z-only gradients:
;gpz6: 100%
;gpz7: -17.13% (spoil)

;use gradient files:
;gpnam6: SINE.100
;gpnam7: SINE.100

;use AU-program dosy to calculate gradient-file Difframp

;$Id: stebpgp1s,v 1.4.6.1 2007/05/09 09:36:59 ber Exp $

error: pathspec 'stebpgp_f2dec.cw' did not match any file(s) known to git


adding sfhmqcf3et.nuws.cw

adding sfhmqcf3et.nuws.cw

;sfhmqcf3et.nuws.cw
;1H,15N SOFAST HMQC
;phase sensitive using Echo/Anti-Echo (Hurd & John 1991)
; Modern Instrumental Analysis, edited by Satinder Ahuja, Neil Jespersen, p286
;with apodisation-weighted sampling
;with decoupling during acquisition
;
;set sampling schedule via vclist:
;  total number of scans = ns * c
;  add one to first point of vclist to allow for dummy scans
;
;modified Chris Waudby 02/05/18
;
;P.Schanda and B. Brutscher, J. Am. Chem. Soc. 127, 8014 (2005)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

#ifdef NUWS
define loopcounter myCounter
"myCounter=0"
#endif /* NUWS */

"cnst4=92"

"d11=30m"
"d12=20u"
"d21=1s/(cnst4*2)"

"p22=2*p21"

"in0=inf1/2"

"d0=3u"

"DELTA=p21*0.6366+p17+d17-p40*0.5-d0"
"DELTA1=d21-p39*cnst39"
"DELTA2=d21-p17-d16-4u"
"acqt0=0"

"spoff23=bf1*(cnst19/1000000)-o1"
"spoff24=bf1*(cnst19/1000000)-o1"

1 ze
  d11 pl26:f3
2 d1 do:f3
3 d12 pl3:f3
  50u UNBLKGRAD

  ; purge equilibrium 15N
  (p21 ph1):f3
  4u
  p16:gp3
  d16

  (p39:sp23 ph1):f1
  DELTA1

  (p21 ph3):f3
  p17:gp1
  d17
  (p22 ph2):f3
  DELTA

  d0
  (p40:sp24 ph1):f1
  d0

  DELTA
  (p22 ph1):f3
  p17:gp1
  d17
  (p21 ph4):f3

  DELTA2
  p17:gp2*EA
  d16 pl26:f3
  4u BLKGRAD
  go=2 ph31 cpd3:f3

#ifdef NUWS
  if "myCounter==1" goto 10
  zd
  "myCounter=1"
10 4u
  ; repeat acquisition block according to schedule in vclist
  lo to 2 times c
  3m ivc
#endif /* NUWS */

  d1 do:f3 mc #0 to 2
     F1EA(igrad EA, id0 & ip3*2 & ip31*2)
exit

ph1=0
ph2=0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
ph3=0 2
ph4=0 0 2 2
ph31=0 2 2 0 2 0 0 2

;pl3 : f3 channel - power level for pulse (default)
;pl26: f3 channel - power level for CPD/BB decoupling (low power)
;sp23: f1 channel - shaped pulse 120 degree
;                   (Pc9_4_120.1000 or Q5.1000)
;sp24: f1 channel - shaped pulse 180 degree (Rsnob.1000)
;p16: homospoil/gradient pulse                       [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p39: f1 channel - 120 degree shaped pulse for excitation
;                      Pc9_4_120.1000 (120o)    (3.0ms at 600.13 MHz)
;                  (or Q5.1000 (90o)            (2.0ms at 600.13 MHz) )
;p40: f1 channel - 180 degree shaped pulse for refocussing
;                      Rsnob.1000               (1.0ms at 600.13 MHz)
;d0 : incremented delay (2D) = in0/2-p3*4/3.1415
;d1 : relaxation delay
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21 : 1/(2J)NH
;cnst4: = J(NH) [92 Hz]
;cnst19: H(N) chemical shift (offset, in ppm)
;cnst39: compensation of chemical shift evolution during p39
;           Pc9_4_120.1000: 0.529
;inf1: 1/SW(N) = 2 * DW(N)
;in0: 1/ SW(N) = 2 * DW(N)
;nd0: 1
;NS: 4 * n
;DS: 16
;aq: <= 50 msec
;td1: number of experiments
;FnMODE: Echo-AntiEcho
;cpd3: decoupling according to sequence defined by cpdprg3: garp4.p62
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence
;          use pulse of >= 350 usec

;for z-only gradients:
;gpz1: 40%
;gpz2: -8.1%
;gpz3: 31%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100

                                          ;preprocessor-flags-start
                                          ;preprocessor-flags-end

;Processing

;PHC0(F1): 0
;PHC1(F1): 0
;FCOR(F1): 0.5

error: pathspec 'sfhmqcf3et.nuws.cw' did not match any file(s) known to git


add hmqcnoesyhmqcccgpph.4d.2.cw

add hmqcnoesyhmqcccgpph.4d.2.cw

;4D 13C HMQC-NOESY-13C HMQC
;gradient selection in second HMQC
;for methyl-methyl NOES
;NUS with random quadrature detection
; (thanks to Daniel Nietlispach)
;Chris W, Dec 2017
;
; uses vclist and vplist to set evolution times and phases
; vclist = t1
;          t2
;          t3
;          ...
; vplist = 1 (r)
;          2 (i)
;          1 (E/AE)
;          ...
;
;F1(H) -> F2(H[mq],t1,d30) -> F2(C[mq],t2,d0) ---NOE--> F1(H) -> F2(C[mq],t3,d10) -> F1(H,t4)
;
;MQ evolution for 1H (taking advantage of methyl trosy)
;Uses half-dwell first-point delay by default in all indirect dims
;Option for off-res presat
;Removal of 13C equilibrium magnetisation
;Delays adjusted for zero first-order phase correction in acqusition dim

;$CLASS=HighRes
;$DIM=4D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

;------------standard delays/pulses
"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d11=30m"

;------------summary
;               t1      t2      t3      t4
;nucleus        1H      13C     13C     1H
;               noe     noe     dir     dir
;axis           F1      F2      F3      F4
;delay          d0      d10     d20
;increments     id0     id10    id20
;initial dwell  1/2     1/2     1/2
;mode           states  states  E/AE    direct
;tppi?          tppi    tppi    tppi
;aqseq 4321

;------------increments
"in0=inf1/2"  ; t1 (1Hnoe)
"in10=inf2/2"  ; t2 (13Cnoe)
"in20=inf3/2"  ; t3 (13Cdir)

;------------initial evolution times
"d0=in0/2-p3"  ; t1 (1Hnoe)
"d10=in10/2-p2"  ; t2 (13Cnoe)
"d20=3u"  ; t3 (13Cdir)

;------------other delays
"TAU=d8-p16*2-d16*2-p3"
"DELTA=p3*0.6366+p17+d17-p2-d0"
"DELTA1=d2-p16-d16"
"DELTA2=d2-0.6366*p1"
"DELTA3=d2-p17-d16-4u-de"

"acqt0=de"

;------------NUS preamble
; loop counters
"l5=1" ; t1
"l6=1" ; t2
"l7=1" ; t3
"l0=1" ; for gradient selection in t3

;set l1 = numner of NUS points to record

;------------loop for NUS points
ze
1 d11 pl12:f2

;------------calculate NUS stuff here
; 1. reset loop counters and phases
20u rp11
20u rp12
20u rp13
20u rp31
20u ru5
20u ru6
20u ru7

; 2. set l5,l6,l7 = vc
41 20u iu5
  lo to 41 times c
  20u ivc

51 20u iu6
  lo to 51 times c
  20u ivc

61 20u iu7
  lo to 61 times c
  20u ivc

"d31=20m-20u*(l5-1)-20u*(l6-1)-20u*(l7-1)"
d31 ; compensate for calculation time

; 3. states-tppi for phases p11,p12,p13
if "l5%2 == 1"
  {
  ip11*2
  ip31*2
  }
if "l6%2 == 1"
  {
  ip12*2
  ip31*2
  }
if "l7%2 == 1"
  {
  ip13*2
  ip31*2
  }

; 4. set phases for quadrature detection
if "vp > 1"  ; t1
  {
  ip11
  }
  20u ivp

if "vp > 1"  ; t2
  {
  20u ip12
  }
  20u ivp

if "vp == 1"  ; t3 (echo/anti-echo)
  {
  "l0=1"
  }
else
  {
  "l0=2"
  }
  20u ivp

; 5. calculate evolution times
20u rd0
20u rd10
20u rd20
"d0=d0+in0*(l5-2)"
"d10=d10+in10*(l6-2)"
"d20=d20+in20*(l7-2)"
20u

;------------ns loop
2 d11 do:f2
  4u BLKGRAD

# ifdef OFFRES_PRESAT
    30u fq=cnst21(bf hz):f1
# endif /*OFFRES_PRESAT*/
  4u pl9:f1
  d1 cw:f1 ph29
  4u do:f1
  4u pl1:f1 pl2:f2
  4u fq=0:f1
  4u UNBLKGRAD

;------------kill equm 13C magnetisation
  (p3 ph1):f2
  4u
  p16:gp1
  d16*2

;------------start first 13C HMQC element
  (p1 ph11):f1
  DELTA1
  p16:gp2
  d16

  (p3 ph12):f2
  ; t1 - 1H F1 indirect evolution (as MQ)
  d0
  (p4 ph2):f2
  d0
  ; t2 - 13C F2 evolution (MQ)
  d10
  (p1 ph1):f1
  (p2 ph2):f1
  (p1 ph1):f1
  d10
  (p3 ph1):f2

  p16:gp2
  d16
  DELTA1
  (p1 ph3):f1

;------------NOE mixing period
  TAU

  p16:gp3*0.71
  d16
  (p3 ph1):f2
  p16:gp3
  d16

;------------start second 13C HMQC element (gradient selected)
  (p1 ph1):f1
  DELTA2

  (p3 ph13):f2
  p17:gp4
  d17
  (p4 ph2):f2
  DELTA
  d20
  (p1 ph1):f1
  (p2 ph2):f1
  (p1 ph1):f1
  d20
  DELTA
  (p4 ph1):f2
  p17:gp4
  d17
  (p3 ph4):f2

  DELTA3 pl12:f2
  if "l0 % 2 == 1"
    {
    p17:gp5
    d16
    }
  else
    {
    p17:gp5*-1
    d16
    }
  4u BLKGRAD

;------------acquisition (ns)
  go=2 ph31 cpd2:f2

;------------loop back and calculate next NUS/RQC point
  d11 do:f2 wr #0 if #0 zd
  lo to 1 times l1

  20u BLKGRAD

exit

ph1= 0
ph2= 1
;ph3=(8) 1
ph3= 0
ph11=0
ph12=0 2
ph13=0
ph29=0
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl9 : f1 channel - power level for presaturation
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse                       [1 msec]
;p17: second gradient pulse                          [250 usec]
;d0 : incremented delay (4D)
;d10: incremented delay (4D)
;d20: incremented delay (4D)
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)CH
;d8 : mixing time
;d11: delay for disk I/O                             [30 msec]
;d16: delay for homospoil/gradient recovery [200 usec]
;d17: short delay for homospoil/gradient recovery [100 usec]
;l1: number of NUS points
;cnst2: = J(CH)
;cnst21: frequency in Hz for off-res presat
;inf1: 1/SW(H)
;inf2: 1/SW(C) = 2 * DW(C)
;inf3: 1/SW(C) = 2 * DW(C)
;in0: 1/(2 * SW(H)) = DW(H)
;in10: 1/(2 * SW(C)) = DW(C)
;in20: 1/(2 * SW(C)) = DW(C)
;nd0: 2
;nd10: 2
;nd20: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments in F1
;td2: number of experiments in F2
;td3: number of experiments in F3
;FnMODE: States-TPPI (or States) in F1
;FnMODE: States-TPPI (or States) in F2
;FnMODE: States-TPPI (or States) in F3
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 31%
;gpz2: 13%
;gpz3: 50%
;gpz4: 40%
;gpz5: 20.1%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SINE.20
;gpnam5: SINE.20

                                          ;preprocessor-flags-start
;OFFRES_PRESAT: for off-resonance presaturation, set cnst21=o1(water)
;F2_plane: for zero 13C phase evolution in F3
;F3_plane: for zero 13C phase evolution in F2
;NUS: for non-uniform sampling (Topspin 3)
                                          ;preprocessor-flags-end

error: pathspec 'hmqcnoesyhmqcccgpph.4d.2.cw' did not match any file(s) known to git


add sfhzdqcnoesysfhmqcgpph.cw

add sfhzdqcnoesysfhmqcgpph.cw

;3D methyl C(Z/D-quantum)CH SFH(Z/D)QC-NOESY-SFHMQC
;for methyl-methyl NOES
;run with F1 as QF, td = 4 * actual td (for interleaving)

;Option for NUS using Topspin 3
;Chris W, Oct 2018

;F1(H) -> F2(C[ZQ/DQ],t1) ---NOE--> F1(H) -> F2(C[mq],t2) -> F1(H,t3)
;
;Indirect evolution order is t2, t1 (13Cdir, 13Cnoe)
;Uses half-dwell first-point delay by default in all indirect dims
;Option for off-res presat
;Options for 2D planes in each 13C dim
;  (set both to get 1D or 2D HH plane with no 13C phase evolution)
;Removal of 13C equilibrium magnetisation
;Delays adjusted for zero first-order phase correction in acqusition dim

;$CLASS=HighRes
;$DIM=4D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d11=45m" ; for disk access

;------------options for first (in transfer pathway) 13C dim (F1)
"in0=inf1"		; first 13C dim
"d0=in0/2-p3*4/3.1415"

;------------options for second (in transfer pathway) 13C dim (F2)
"in10=inf2"		; second 13C dim
"d10=in10/2-p3*4/3.1415"

; place pulses on-resonance
;"spoff23=bf1*(cnst19/1000000)-o1"
;"spoff24=bf1*(cnst19/1000000)-o1"
;"spoff25=bf1*(cnst19/1000000)-o1"

"TAU=d8-p16*2-d16*2-p3-8u"  ; noe mixing time

; delays for first SFHMQC
"DELTA1=d2-p41*cnst39-p16-d16"
"DELTA4=d2-p41*cnst39-p16-d16-p40*0.5"

; delays for second SFHMQC
"DELTA2=d2-p16-d16-p39*cnst39"
"DELTA3=p39*cnst39-de-4u"
"acqt0=de"

aqseq 312

1 ze
  d11 pl12:f2
  4u BLKGRAD
2 d11 do:f2
  d1 pl2:f2
  50u UNBLKGRAD

;-------------------------kill equm 13C magnetisation

  (p3 ph1):f2
  4u
  p16:gp1
  d16*2

;-------------------------start first 13C HMQC element
;uses DELTA1, DELTA4, DELTA5
  (p41:sp25 ph11):f1
  p16:gp2
  d16
  (lalign (DELTA4 p40:sp24 ph1) (DELTA1 p3 ph12 d0 p3 ph1 DELTA1):f2)
  p16:gp2
  d16
  (p41:sp25 ph1):f1

;------------------------start NOE period
  4u
  p16:gp3*0.71
  d16
  (p3 ph1):f2
  4u
  p16:gp3
  d16

  TAU

;------------------------start second 13C HMQC element (120o excitation)
;uses DELTA2, DELTA3
  (p39:sp23 ph1):f1
  p16:gp4
  d16

  (center (p40:sp24 ph2):f1 (DELTA2 p3 ph13 d10 p3 ph1 DELTA2):f2 )

  DELTA3
  p16:gp4
  d16 pl12:f2
  4u BLKGRAD
  go=2 ph31 cpd2:f2

  d11 do:f2 mc #0 to 2
    F1I(ip11, 2, ip12, 2)
    F1QF(id0)
    F2PH(rd0 & ip13, id10)
  4u BLKGRAD

exit

ph1= 0
ph2= 1
ph11=0
ph12=0
ph13=0 2
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl9 : f1 channel - power level for presaturation
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse                       [1 msec]
;p19: second gradient pulse                          [250 usec]
;p22: f3 channel - 180 degree high power pulse
;p28: f1 channel - trim pulse                        [1 msec]
;d0 : incremented delay (4D)
;d10: incremented delay (4D)
;d20: decremented delay (4D)
;d28: incremented delay (4D)
;d30: incremented delay (4D)
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)CH
;d8 : mixing time
;d11: delay for disk I/O                             [45 msec]
;d12: delay for power switching                      [20 usec]
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(CH)
;cnst21: frequency in Hz for off-res presat
;inf1: 1/SW(H)
;inf2: 1/SW(C) = 2 * DW(C)
;inf3: 1/SW(C) = 2 * DW(C)
;in0: 1/(2 * SW(C)) = DW(C)
;in10: 1/(2 * SW(C)) = DW(C)
;in30: 1/(2 * SW(H)) = DW(H)
;nd0: 2
;nd10: 2
;NS: 2 * n
;DS: 128
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or States) in F1
;FnMODE: States-TPPI (or States) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;p39: f1 channel - 120 degree shaped pulse for excitation
;                      Pc9_4_120.1000 (120o)
;p40: f1 channel - 180 degree shaped pulse for refocussing
;                      Rsnob.1000  /  Reburp
;p41: f1 channel - 90 degree shaped pulse for excitation
;                      Pc9_4_90.1000
;cnst39: compensation of chemical shift evolution during p39
;           Pc9_4_120.1000: 0.529

;for z-only gradients:
;gpz1: 31%
;gpz2: 29%
;gpz3: 23%
;gpz4: 13%
;gpz5: 43%
;gpz6: 19%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SINE.50
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.50
;gpnam6: SINE.100

                                          ;preprocessor-flags-start
;NUS: for non-uniform sampling (Topspin 3)
                                          ;preprocessor-flags-end

error: pathspec 'sfhzdqcnoesysfhmqcgpph.cw' did not match any file(s) known to git


created b_trosy_hmqc.cw

created b_trosy_hmqc.cw

;BEST-TROSY-HMQC
;Chris Waudby, Aug 2019
;
;options:
; -DONE_D = first-row
; -DOFFRES_PRESAT = presat, pl9 on cnst21 (Hz bf)

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"d11=30m"
"d12=20u"
"d13=4u"
"d21=1s/(cnst4*4)"

"p22=p21*2"

"in0=inf1"
# ifdef ONE_D
"d0=2u"
#else
"d0=in0/2-p21*4/3.1415"
# endif /*ONE_D*/

"d2=p39-p39*cnst39-0.3633*p21"
"d3=0.5*p40-0.3633*p21"
"DELTA1=d21-p39*cnst39-p40*0.5-p16-d16-4u"
"DELTA2=d21-0.3633*p21-p16-d16-4u-0.5*p40"
"DELTA3=d21-p40-p16-d16-4u"
"DELTA4=d21-0.5*p40-p16-d16-4u-p21-de"
"acqt0=de"

#ifdef LABEL_CN
"d10=DELTA3+d3+p21+d0*0.5-p8*0.5"
"d9=DELTA3+d3+p21+d0*0.5-p8*0.5"
"in10=in0*0.5"
"in9=in0*0.5"
#endif

# ifdef OFFRES_PRESAT
  "TAU=d1-d11-60u-d12*2-d13-d12-50u-p21-2*p16-2*d16-12u"
# else
  "TAU=d1-d11-d12-50u-p21-2*p16-2*d16-12u"
# endif /*OFFRES_PRESAT*/

;"spoff23=bf1*(cnst19/1000000)-o1"
;"spoff24=bf1*(cnst19/1000000)-o1"
"spoff23=0" ; for amides on-resonance (recommended)
"spoff24=0"

"l0=1"  ; loop counter for shifting 1H 180 pulse between echo/anti-echoes

1 ze
  d11
2 d11

  4u UNBLKGRAD
  p16:gp3
  d16
  4u BLKGRAD

# ifdef OFFRES_PRESAT
  30u fq=cnst21(bf hz):f1
  d12 pl9:f1
  TAU cw:f1 ph1
  d13 do:f1
  d12 pl1:f1
  30u fq=0:f1
# else
  TAU
# endif /*OFFRES_PRESAT*/

3 d12 pl3:f3
  50u UNBLKGRAD

  ; purge Nz
  (p21 ph1):f3
  4u
  p16:gp0
  d16

  ; begin main sequence
  if "l0 %2 == 1"
     {
  ; echo
  (p39:sp23 ph10) (d2 p21 ph11):f3
  DELTA1
  4u
  p16:gp1
  d16
  (center (p40:sp24 ph1) (p22 ph11):f3 )
  4u
  p16:gp1
  d16
  (center (p40:sp24 ph1):f1 (DELTA2 p21 ph12 d0 p21 ph13 DELTA2):f3 )
  4u
  p16:gp2
  d16
  (center (p40:sp24 ph1) (p22 ph13):f3 )
  4u
  p16:gp2
  d16
  DELTA4 BLKGRAD
  (p21 ph14):f3
     }
  else
     {
  ; anti-echo
  (p39:sp23 ph10) (d2 p21 ph21):f3
  DELTA1
  4u
  p16:gp1
  d16
  (center (p40:sp24 ph1) (p22 ph21):f3 )
  4u
  p16:gp1
  d16
  (center (p40:sp24 ph1):f1 (DELTA2 p21 ph22 d0 p21 ph23 DELTA2):f3 )
  4u
  p16:gp2
  d16
  (center (p40:sp24 ph1) (p22 ph23):f3 )
  4u
  p16:gp2
  d16
  DELTA4 BLKGRAD
  (p21 ph24):f3
     }
  go=2 ph31
  d11 mc #0 to 2
     F1EA(iu0, id0)

exit

ph1=0
ph10=0

;spin dynamic phases:
;ph11=0 1 2 3  3 0 1 2  2 3 0 1  1 2 3 0
;ph12=1 2 3 0  0 1 2 3  3 0 1 2  2 3 0 1
;ph13=0 1 2 3
;ph14=1 2 3 0
;ph21=0 1 2 3  1 2 3 0  2 3 0 1  3 0 1 2
;ph22=3 0 1 2  0 1 2 3  1 2 3 0  2 3 0 1
;ph23=0 1 2 3
;ph24=3 0 1 2

;bruker phases:
ph11=0 1 2 3
ph12=1 0 3 2
ph13=0
ph14=3
ph21=0 3 2 1
ph22=3 0 1 2
ph23=0
ph24=1

ph31=0 1 2 3

;pl3 : f3 channel - power level for pulse (default)
;pl26: f3 channel - power level for CPD/BB decoupling (low power)
;sp23: f1 channel - shaped pulse 120 degree
;                   (Pc9_4_120.1000 or Q5.1000)
;sp24: f1 channel - shaped pulse 180 degree (Rsnob.1000)
;p16: homospoil/gradient pulse                       [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p39: f1 channel - 120 degree shaped pulse for excitation
;                      Pc9_4_120.1000 (120o)    (3.0ms at 600.13 MHz)
;                  (or Q5.1000 (90o)            (2.0ms at 600.13 MHz) )
;p40: f1 channel - 180 degree shaped pulse for refocussing
;                      Rsnob.1000               (1.0ms at 600.13 MHz)
;d0 : incremented delay (2D) = in0/2-p21*4/3.1415
;d1 : relaxation delay
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21 : 1/(4J)NH
;cnst4: = J(NH)
;cnst19: H(N) chemical shift (offset, in ppm)
;cnst39: compensation of chemical shift evolution during p39
;           Pc9_4_90.1000: 0.514
;           Pc9_4_120.1000: 0.529
;inf1: 1/SW(N) = 2 * DW(N)
;in0: 1/ SW(N) = 2 * DW(N)
;nd0: 1
;NS: 4 * n
;DS: 16
;td1: number of experiments
;FnMODE: Echo-AntiEcho

;use gradient ratio:	gp 0 : gp 1 : gp 2
;			-16 :  11 :    7

;for z-only gradients:
;gpz0: -16%
;gpz1: 11%
;gpz2:  7%
;gpz3: -23%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

error: pathspec 'b_trosy_hmqc.cw' did not match any file(s) known to git


adding sfhmqcf3gpph_T2.cw (tested)

adding sfhmqcf3gpph_T2.cw (tested)

;Added option for off-resonance presat (e.g. to suppress urea signal), 21/6/15
;
;With option for 1D (first row)
;
;sfhmqcf3gpph
;avance-version (09/11/18)
;SOFAST HMQC
;2D H-1/X correlation via heteronuclear zero and double quantum
;   coherence
;phase sensitive
;with decoupling during acquisition
;
;P.Schanda and B. Brutscher, J. Am. Chem. Soc. 127, 8014 (2005)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst4=92"

"p22=2*p21"
"d11=30m"
"d12=20u"
"d13=4u"
"d21=1s/(cnst4*2)"

"in0=inf2"

# ifndef ONE_D

"d0=in0/2-p21*4/3.1415"

# endif /*ONE_D*/

"DELTA=d21-p16-d16"
"DELTA4=d16-de-4u"
"acqt0=de"

# ifdef OFFRES_PRESAT
  "TAU=d1-10m-60u-d12*2-d13"
# else
  "TAU=d1-10m"
# endif /*OFFRES_PRESAT*/

define delay vdmin
"vdmin=4*larger(p21+0.5*p40,p21+cnst39*p39)"

"spoff23=bf1*(cnst19/1000000)-o1"
"spoff24=bf1*(cnst19/1000000)-o1"

1 ze
  d11 pl26:f3
2 10m do:f3

  "DELTA1=0.25*vd-p21-cnst39*p39"
  "DELTA2=0.25*vd-p21-0.5*p40"
  "DELTA3=0.25*vd-p21"

# ifdef OFFRES_PRESAT

  30u fq=cnst21(bf hz):f1
  d12 pl9:f1
  TAU cw:f1 ph29
  d13 do:f1
  d12 pl1:f1
  30u fq=0:f1

# else

  TAU

# endif /*OFFRES_PRESAT*/

3 d12 pl3:f3
  50u UNBLKGRAD

  p16:gp2
  d16

  (p39:sp23 ph1):f1
  DELTA1
  (p22 ph1):f3
  DELTA2
  (p40:sp24 ph13):f1
  DELTA2
  (p22 ph2):f3
  DELTA3

  p16:gp1
  d16

# ifndef ONE_D

#   ifdef LABEL_CN
  (center (p40:sp24 ph2):f1 (p8:sp13 ph1):f2 (DELTA p21 ph3 d0 p21 ph4 DELTA):f3 )
#   else
  (center (p40:sp24 ph2):f1 (DELTA p21 ph3 d0 p21 ph4 DELTA):f3 )
#   endif /*LABEL_CN*/

# else

 (center (p40:sp24 ph2):f1 (DELTA p21 ph3 6u p21 ph4 DELTA):f3 )

# endif /*ONE_D*/

  p16:gp1
  DELTA4 pl26:f3
  4u BLKGRAD
  go=2 ph31 cpd3:f3
  10m do:f3 mc #0 to 2
     F1QF(ivd)
     F2PH(ip3, id0)
exit

ph1=0
ph2=0
ph3=0 2
ph4=0 0 2 2
ph13=0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
ph29=0
ph31=0 2 2 0 2 0 0 2

;pl3 : f3 channel - power level for pulse (default)
;pl26: f3 channel - power level for CPD/BB decoupling (low power)
;sp13: f2 channel - shaped pulse 180 degree (adiabatic)
;sp23: f1 channel - shaped pulse 120 degree
;                   (Pc9_4_120.1000 or Q5.1000)
;sp24: f1 channel - shaped pulse 180 degree (Rsnob.1000)
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p16: homospoil/gradient pulse                       [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p39: f1 channel - 120 degree shaped pulse for excitation
;                      Pc9_4_120.1000 (120o)    (3.0ms at 600.13 MHz)
;                  (or Q5.1000 (90o)            (2.0ms at 600.13 MHz) )
;p40: f1 channel - 180 degree shaped pulse for refocussing
;                      Rsnob.1000               (1.0ms at 600.13 MHz)
;d0 : incremented delay (2D) = in0/2-p21*4/3.1415
;d1 : relaxation delay
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21 : 1/(2J)NH
;cnst4: = J(NH)
;cnst19: H(N) chemical shift (offset, in ppm)
;cnst39: compensation of chemical shift evolution during p39
;           Pc9_4_120.1000: 0.529
;           Q5.1000: -0.07
;inf1: 1/SW(N) = 2 * DW(N)
;in0: 1/ SW(N) = 2 * DW(N)
;nd0: 1
;NS: 2 * n
;DS: 16
;aq: <= 50 msec
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEC
;cpd3: decoupling according to sequence defined by cpdprg3: garp4.p62
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence
;          use pulse of >= 350 usec

;use gradient ratio:	gp 1 : gp 2
;			  11 :    7

;for z-only gradients:
;gpz1: 11%
;gpz2:  7%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100

                                          ;preprocessor-flags-start
;LABEL_CN: for C-13 and N-15 labeled samples start experiment with
;             option -DLABEL_CN (eda: ZGOPTNS)
                                          ;preprocessor-flags-end

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

;$Id: sfhmqcf3gpph,v 1.1.2.8 2009/11/18 11:19:58 ber Exp $

error: pathspec 'sfhmqcf3gpph_T2.cw' did not match any file(s) known to git


add cosy1pulse.cw

add cosy1pulse.cw

;cosyph
;avance-version (12/01/11)
;2D homonuclear shift correlation
;phase sensitive
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>

"in0=inf1"

"d0=3u"
"d2=p1*0.6366+3u"
"acqt0=-d2"

1 ze
2 d1
  p1 ph1
  d0
  go=2 ph31
  d1 mc #0 to 2 F1QF(id0)
exit

ph1= 0 1 2 3
ph31=0 1 2 3

;pl1 : f1 channel - power level for pulse (default)
;p0 : f1 channel -  20 to 90 degree high power pulse
;p1 : f1 channel -  90 degree high power pulse
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;inf1: 1/SW = 2 * DW
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

;$Id: cosyph,v 1.9 2012/01/31 17:49:22 ber Exp $

error: pathspec 'cosy1pulse.cw' did not match any file(s) known to git


creating b_hsqcf3gpph.cw

creating b_hsqcf3gpph.cw

;BEST-15N HSQC, using PC-9 shapes for all 90deg pulses on Hn
;Chris W, Aug 2018
;
;b_hsqcf3gpph.cw
;avance-version (08/01/24)
;(best)-HSQC
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition
;using f3 - channel
;with gradients in back-inept
;
;(A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn.
;   Reson. 93, 151-170 (1991) )
;(L.E. Kay, P. Keifer & T. Saarinen, J. Am. Chem. Soc. 114,
;   10663-5 (1992) )
;(J. Schleucher, M. Schwendinger, M. Sattler, P. Schmidt,
;   O. Schedletzky, S.J. Glaser, O.W. Sorensen & C. Griesinger,
;   J. Biomol. NMR 4, 301-306 (1994) )
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p22=p21*2"
"d11=30m"
"d12=20u"
"d26=1s/(cnst4*4)"

"d0=inf1/2-2*0.6366*p21"
"in0=inf1"

"DELTA1=d26-p19-d16-p41*cnst41-p42*cnst42"
"DELTA2=DELTA1+p41*cnst41-4u-de"

"spoff25=bf1*(cnst19/1000000)-o1"
"spoff26=bf1*(cnst19/1000000)-o1"
"spoff27=bf1*(cnst19/1000000)-o1"
"d42=p42"

1 ze
  d11 pl26:f3 pl0:f1
2 d11 do:f3
  d1
  50u UNBLKGRAD
  4u pl3:f3

  ; first INEPT
  (p41:sp25 ph1)
  DELTA1
  p19:gp1
  d16
  (center (p42:sp26 ph1) (p22 ph1):f3 )
  DELTA1
  p19:gp1
  d16
  (p41:sp27 ph2):f1

  ; zz filter
  p16:gp2
  d16

  ; t1 evolution
#   ifdef LABEL_CN
  (center
    (p42:sp26 ph5):f1
    (p8:sp13 ph1):f2
    (d42*0.5 p21 ph3 d0 p21 ph4 d42*0.5):f3
  )
#   else
  (center
    (p42:sp26 ph5):f1
    (d42*0.5 p21 ph3 d0 p21 ph4 d42*0.5):f3
  )
#   endif /*LABEL_CN*/

  ; zz filter
  p16:gp3
  d16

  ; retro-INEPT
  (p41:sp25 ph1)
  DELTA1
  p19:gp1
  d16
  (center (p42:sp26 ph1) (p22 ph1):f3 )
  DELTA2 pl26:f3
  p19:gp1
  d16
  4u BLKGRAD

  ; acquisition
  go=2 ph31 cpd3:f3
  d11 do:f3 mc #0 to 2
     F1PH(ip3, id0)
exit

ph1=0
ph2=1
ph3=0 2
ph4=0 0 2 2
ph5=0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
ph31=0 2 2 0

;pl3 : f3 channel - power level for pulse (default)
;pl26: f3 channel - power level for CPD/BB low power decoupling
;sp13: f2 channel - shaped pulse 180 degree (adiabatic)
;sp25: f1 channel - shaped pulse  90 degree (Pc9_4_90.1000)
;sp26: f1 channel - shaped pulse 180 degree (Reburp.1000)
;sp27: f1 channel - shaped pulse  90 degree (Pc9_4_90.1000)
;                   for time reversed pulse
;sp30: f1 channel - shaped pulse 180 degree (Bip720,50,20.1)
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p16: homospoil/gradient pulse                         [1 msec]
;p19: gradient pulse 2                                 [500 usec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p29: gradient pulse 3                                 [250 usec]
;p41: f1 channel -  90 degree shaped pulse for excitation
;                      Pc9_4_90.1000             (2230 usec at 600 MHz)
;p42: f1 channel - 180 degree shaped pulse for refocusing
;                      Reburp.1000               (1680 usec at 600 MHz)
;p43: f1 channel -  90 degree shaped pulse for excitation
;                      Bip720,50,20.1            (200us at 600 MHz)
;d0 : incremented delay (2D)                           [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                               [30 msec]
;d16: delay for homospoil/gradient recovery
;d26: 1/(4J(NH))
;d63: set to zero unless DELTA becomes negative
;cnst4: = J(NH)
;cnst19: H(N) chemical shift (offset, in ppm)
;cnst41: compensation of coupling evolution during p41
;           Pc9_4_90.1000: 0.514
;cnst42: compensation of coupling evolution during p42
;           Reburp.1000: 0.475
;NS: 2 * n
;DS: >= 16
;aq: <= 50 msec (or <100ms with d1>100ms)
;td1: number of experiments
;FnMODE: echo-antiecho
;cpd3: decoupling according to sequence defined by cpdprg3: garp4.p62
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 8%
;gpz2: 13% for N-15
;gpz3: -7%

;use gradient files:
;gpnam1: SMSQ10.50
;gpnam2: SMSQ10.50
;gpnam3: SMSQ10.50

                                          ;preprocessor-flags-start
;LABEL_CN: for C-13 and N-15 labeled samples start experiment with
;             option -DLABEL_CN (eda: ZGOPTNS)
                                          ;preprocessor-flags-end

error: pathspec 'b_hsqcf3gpph.cw' did not match any file(s) known to git


creating noesyhsqcfpf3gpphwg.2.cw

creating noesyhsqcfpf3gpphwg.2.cw

;NOESY-15N HSQC (derived from hsqcfpf3gpphwg.3.2.jk)
;For 15N-labelled samples only (no 13C decoupling)
;With water flipback
; using separate pulses for initial 1H evolution and in HSQC

;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"p2=p1*2"
"p22=p21*2"
"d11=30m"
"d12=20u"
;"d13=4u"
"d26=1s/(cnst4*4)"

"in0=inf1"
"in10=inf2/4"

"DELTA1=d26-p16-d16-4u"
"DELTA2=d26-p19-d16-p11-d12-4u"
"DELTA3=d26-p19-d16-p11-d12-8u-de+0.63662*p1"

"TAU=d8-p16-d16-p11-20u"

# ifdef H_SINGLEDWELL
    "d0=in0-1.27324*p1"
# else
    "d0=in0/2-1.27324*p1"
# endif /*H_SINGLEDWELL*/

# ifdef N_SINGLEDWELL
    "d10=in10-0.5*(10u+p1+0.63662*p21)"
# else
    "d10=in10/2-0.5*(10u+p1+0.63662*p21)"
# endif /*N_SINGLEDWELL*/

"acqt0=0"

aqseq 321

1 ze
  d11 pl16:f3
2 d11 do:f3

  d1

3 d12 pl1:f1 pl3:f3
  50u UNBLKGRAD

  (p21 ph1):f3
  4u
  p16:gp1
  d16
  (p21 ph2):f3
  4u
  p16:gp1*0.7
  d16*2

# ifndef NH_PLANE

    10u pl0:f1
    (p12:sp2 ph20:r):f1		; flipdown(-x/x), z -> y/-y
    10u pl1:f1

    if "d0+2*p1 < p22"
    {
      (p1 ph10 d0 p1 ph11):f1
    }
    else
    {
      (center (p1 ph10 d0 p1 ph11):f1 (p22 ph5):f3 )
    }

    10u pl0:f1
    (p12:sp12 ph21:r):f1	; flipback(-x/x), -y/y -> z
    10u pl1:f1

    TAU

    p16:gp2
    d16

# endif /*NH_PLANE*/

  (p1 ph1)
  4u
  p16:gp3
  d16
  DELTA1
  (center (p2 ph2) (p22 ph6):f3 )
  DELTA1
  4u
  p16:gp3
  d16
  (p1 ph2)

  4u pl0:f1
  (p11:sp1 ph8:r):f1	; flipdown(x), -y -> -z
  4u
  p16:gp4
  d16 pl1:f1

# ifdef HH_PLANE

    (p21 ph3):f3
    2u
    (p21 ph4):f3
    2u
    (p2 ph5):f1

# else

    (p21 ph3):f3
    2u
    d10 gron0
    d10 gron0*-1
    8u groff
    (p2 ph5):f1
    2u
    d10 gron0
    d10 gron0*-1
    8u groff
    (p21 ph4):f3

# endif /*HH_PLANE*/

  4u
  p16:gp6
  d16 pl0:f1
  (p11:sp1 ph7:r):f1	; flipdown(-x), z -> y
  4u
  4u pl1:f1

  (p1 ph1)
  4u
  p19:gp5
  d16 pl0:f1
  DELTA2
  (p11:sp1 ph7:r):f1	; flipdown(-x), z -> y
  d12 pl1:f1
  (center (p2 ph1) (p22 ph1):f3 )
  d12 pl0:f1
  (p11:sp11 ph9:r):f1	; flipback(-x), -y -> z
  4u
  p19:gp5
  d16 pl16:f3
  DELTA3
  4u BLKGRAD

  go=2 ph31 cpd3:f3
  d11 do:f3 mc #0 to 2
	F1PH(rp3 & rp6 & rd10 & ip11 & ip21, id0)
	F2PH(ip3 & ip6, id10)
exit

ph1=0
ph2=1
ph6=0
ph7=2
ph8=0
ph9=2

# ifndef NH_PLANE

ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph10=0 0 2 2
ph11=0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
ph20=2 2 0 0
ph21=2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
ph31=0 2 2 0 2 0 0 2 0 2 2 0 2 0 0 2
     2 0 0 2 0 2 2 0 2 0 0 2 0 2 2 0

# else

ph3=0 2
ph4=0 0 2 2
ph5=0 0 0 0 2 2 2 2
ph31=0 2 2 0

# endif /*NH_PLANE*/

;pl0 : 120dB
;pl1 : f1 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl16: f3 channel - power level for CPD/BB decoupling
;sp1: f1 channel - shaped pulse  90 degree (flip-down)
;sp11: f1 channel - shaped pulse  90 degree (flip-back)
;sp2: f1 channel - shaped pulse  90 degree (flip-down) (more selective)
;sp12: f1 channel - shaped pulse  90 degree (flip-back) (more selective)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p11: f1 channel -  90 degree shaped pulse  [1000 usec]
;p12: f1 channel -  90 degree shaped pulse, more selective  [4000 usec]
;p16: homospoil/gradient pulse
;p19: second homospoil/gradient pulse
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;d10 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d8 : mixing time
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d26 : 1/(4J)NH
;cnst4: = J(NH)
;inf1: 1/SW(H) = 4 * DW(H)
;inf2: 1/SW(N) = 4 * DW(N)
;in0: 1/(2 * SW(H)) = DW(H)
;in10: 1/(2 * SW(N)) = DW(N)
;nd0: 4
;nd10: 4
;NS: 4 * n
;DS: 16
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI, TPPI, States or QSEQ in F1
;FnMODE: States-TPPI, TPPI, States or QSEQ in F2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz0: 1-2%
;gpz1: 47%
;gpz2: 11%
;gpz3: 7%
;gpz4: 17%
;gpz5: 53%
;gpz6: 13%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
;gpnam6: SMSQ10.100

error: pathspec 'noesyhsqcfpf3gpphwg.2.cw' did not match any file(s) known to git


create QQ_CPMG (untested)

create QQ_CPMG (untested)

/*  QQ CPMG (Chris Waudby Aug 2021)
    based on 13CH3_1H_TQ_CPMG_ty_800_cpz

    This pulse sequence will allow one to perform the following experiment:

    2D 1H/13C to measure exchange via 13CH3 CPMG

    Assumes that sample is specifically 13CH3 labeled

        1H: O1 on methyls (0.8 ppm)
            pwh = p1 1H pw90 @ power level pl1 highest power
            pwh_cp = p15 1H pw90 @ power level pl15 for CPMG pulses

       13C: O2 centre at 20 ppm
            pwc = p2 13C pw90 @ power level pl2 highest power
            power level pl21 is used for 13C decoupling.

        2H: O4 centre at ~0.8 ppm (methyl 1H resonances)
            pwd = p4 2H pw90 @ power level pl4 high power
            power level pl41 is used for 2H waltz-16 decoupling.
            Note: 2H decoupling only necessary for measuring nuCPMG < 100 Hz

       Recommend: use -Dcomp_flg to apply composite 180o CPMG pulses

       Written by T. Yuwen on Nov 21, 2016
*/

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

/***********************/
/*   Define pulses     */
/***********************/
define pulse dly_pg1     /* Messerle purge pulse */
  "dly_pg1=2m"
define pulse dly_pg2     /* Messerle purge pulse */
  "dly_pg2=3.4m"
define pulse pwh
  "pwh=p1"               /* 1H hard pulse at power level p1 (tpwr) */
define pulse pwc
  "pwc=p2"               /* 13C pulse at power level pl2 (dhpwr) */

#ifdef Ddec
define pulse pwd
  "pwd=p4"               /* 2H pulse at power pl4 */
#endif

define pulse pwh_cp      /* 1H CPMG pulse power level */
  "pwh_cp=p15"

/************************/
/*   Define delays      */
/************************/
"in0=inf2/2"
"d11=30m"

/************************/
/*   Define f1180       */
/************************/
  "d0=larger((in0)/2 - 2.0*pwh_cp, 2e-7)"

define delay taua
  "taua=d3"              /* d3 ~ 1.8-2ms ~ 1.0s/(4*125.3)"  ~ 1 / 4J(CH) */
define delay taub
  "taub=d4"              /* d4 = 1/4JCH exactly */

define delay time_T2
  "time_T2=d6"           /* CPMG duration <= 40 ms  */

/*************************************/
/* Define parameters related to CPMG */
/*************************************/
define delay tauCPMG
define delay tauCPMG1
define delay tauCPMG2

define list<loopcounter> ncyc_cp=<$VCLIST>

"cnst31=1/time_T2"       /* minimum nuCPMG for given time_T2 */

/******************************************************/
/* Assign cnsts to check validity of parameter ranges */
/******************************************************/

"cnst15=plw15"           /* tpwrcp - power level for 1H CPMG pulses */
"cnst21=plw21"           /* dpwr pl21 - set max at 2.0W */

/****************************/
/* Initialize loop counters */
/****************************/
"l1=0"
"l2=0"
"l3=0"

"acqt0=0"              /* select 'DIGIMOD = baseopt' to execute */

aqseq 312

1 ze
#ifdef Ddec
; d11 LOCKDEC_ON /* Not required for AvanceIII-HD */
  50u LOCKH_ON
  d11 H2_PULSE
  2u pl41:f4
#endif /*Ddec*/

2 d11 do:f2
/****************************************/
/* Continue to check run time variables */
/****************************************/
  "l2 = (trunc(ncyc_cp[l1] + 0.3))"

  if "l2 > 80"
  {
    2u
    print "error: ncyc_cp must be < 81"
    goto HaltAcqu
  }

  if "l2 != 0"
  {
    "tauCPMG = time_T2*0.25/l2"
    if "tauCPMG < 124.0u"
    {
      2u
      print "error: tauCPMG < 124u too short for CPMG"
      goto HaltAcqu
    }
  }
  else
  {
    "tauCPMG = time_T2*0.25"
  }

  "tauCPMG1 = tauCPMG - pwh_cp*2.0"
#ifdef comp_flg
  "tauCPMG2 = tauCPMG - pwh_cp*2.0"
#else
  "tauCPMG2 = tauCPMG - pwh_cp"
#endif

  20u pl1:f1 pl2:f2

#ifdef Ddec
  d11 H2_LOCK
  6m LOCKH_OFF
#endif /*Ddec*/

/******************/
/* Messerle purge */
/******************/
  20u pl11:f1
  (dly_pg1 ph26):f1
  20u
  (dly_pg2 ph27):f1
  20u pl10:f1

; off-resonance presat
30u fq=cnst10(bf hz):f1
30u pl10:f1
d1 cw:f1 ph26
4u do:f1
30u fq=0:f1

#ifdef Ddec
  50u LOCKH_ON
  15u H2_PULSE
#endif

/****************************************/
/* Destroy 13C equlibrium magnetization */
/****************************************/
  (pwc ph26):f2

#ifdef Ddec
  20u UNBLKGRAMP
#else
  20u UNBLKGRAD
#endif

  2u
  p50:gp0
  d16

/***********************/
/* Create TQ coherence */
/***********************/
  20u pl15:f1
  2u rpp11 rpp12 rpp13 rpp14 rpp15

  (pwh_cp ph1):f1

  2u
  p51:gp1
  d16

  "DELTA = taua - 2u - p51 - d16 - pwh_cp*2.0/PI"
  DELTA

  (center (pwh_cp*2 ph1):f1 (pwc*2 ph26):f2)

  2u
  p51:gp1
  d16

  "DELTA = taua - 2u - p51 - d16"
  DELTA

  (pwc ph3):f2

  2u
  p52:gp2
  d16

  "DELTA = taub - 2u - p52 - d16"
  DELTA

  (center (pwh_cp*2 ph1):f1 (pwc*2 ph26):f2)

  2u
  p52:gp2
  d16

#ifdef Ddec
  "DELTA = taub - 2u - p52 - d16 - 2u - pwd - 2u - 2u"
#else
  "DELTA = taub - 2u - p52 - d16"
#endif
  DELTA

#ifdef Ddec
  2u pl4:f4
  (pwd ph27):f4
  2u pl41:f4
  (2u cpds4 ph26):f4
#endif

;  (pwc ph26):f2
  (pwh_cp ph1):f1

/*********************************/
/* The first half of CPMG period */
/*********************************/
  if "l2 == 1" {
    tauCPMG1
    (pwh_cp ph29 pwh_cp*2 ph26 pwh_cp ph29):f1
    tauCPMG1
  }

  if "l2 > 1" {
3   tauCPMG2
#ifdef comp_flg
    (pwh_cp ph14 pwh_cp*2 ph11 pwh_cp ph14):f1
#else
    if "l2%2 == 0" {
      (pwh_cp*2 ph15):f1
    }
    else {
      if "abs((l2%8)-4) > 2" {
        (pwh_cp*2 ph13):f1
      }
      else {
        (pwh_cp*2 ph11):f1
      }
    }
#endif
    tauCPMG2 ipp11 ipp12 ipp13 ipp14 ipp15
    lo to 3 times l2
  }

/*****************************/
/* The central 180o 1H pulse */
/*****************************/
  (center (pwh_cp*2 ph2):f1 (pwc*2 ph25):f2 )

/**********************************/
/* The second half of CPMG period */
/**********************************/
  if "l2 == 1" {
    tauCPMG1
    (pwh_cp ph27 pwh_cp*2 ph26 pwh_cp ph27):f1
    tauCPMG1
  }

  if "l2 > 1" {
4   tauCPMG2 dpp11 dpp12 dpp13 dpp14 dpp15
#ifdef comp_flg
    (pwh_cp ph13 pwh_cp*2 ph12 pwh_cp ph13):f1
#else
    if "l2%2 == 0" {
      (pwh_cp*2 ph15):f1
    }
    else {
      if "abs((l2%8)-4) > 2" {
        (pwh_cp*2 ph14):f1
      }
      else {
        (pwh_cp*2 ph12):f1
      }
    }
#endif
    tauCPMG2
    lo to 4 times l2
  }

  (pwh_cp ph26):f1
;  (pwc ph3):f2

#ifdef Ddec
  2u do:f4
  2u pl4:f4
  (pwd ph29):f4
#endif

  2u
  p53:gp3
  d16

#ifdef Ddec
  "DELTA = taub - 2u - 2u - pwd - 2u - p53 - d16"
#else
  "DELTA = taub - 2u - p53 - d16"
#endif
  DELTA

/********/
/* HMQC */
/********/
  (pwc*2 ph26):f2

  d0
  (pwh_cp ph29 pwh_cp*2 ph26 pwh_cp ph29):f1
  d0

  2u
  p53:gp3
  d16

  "DELTA = taub - 2u - p53 - d16"
  DELTA

  (pwc ph4):f2

  "DELTA = pwc*2.0"
  DELTA

  (pwh_cp ph27):f1

  2u
  p54:gp4
  d16

/****************************************************************/
/* C->H back transfer, use wtg_flg for better water suppression */
/****************************************************************/
  20u pl1:f1

  (pwh ph26):f1

  2u
  p57:gp7
  d16
  "DELTA = taua - 2u - p57 - d16 - p10 - 1u - larger(pwh,pwc) - pwh*2.0/PI"
  DELTA
  (p10:sp10 ph28):f1
  1u pl1:f1
  (center (pwh*2 ph26):f1 (pwc*2 ph27):f2 )
  1u
  (p10:sp10 ph28):f1
  "DELTA = taua - p57 - d16 - p10 - 1u - larger(pwh,pwc) - 2*pwc - 8u"
  DELTA
  p57:gp7
  d16

#ifdef Ddec
  4u BLKGRAMP
#else
  4u BLKGRAD
#endif

  (pwc ph26):f2
  (pwc ph5):f2

  4u pl21:f2             /* lower power for 13C decoupling */

/********************************/
/* Signal detection and looping */
/********************************/
  go=2 ph31 cpds2:f2
  d11 do:f2 mc #0 to 2
    F1QF(iu1)
    F2PH(ru1 & ip4, id0); & ip31*2)
;    F1QF(calclc(l1,1))
;    F2PH(calph(ph4,-90), caldel(d0,+in0) & calph(ph31,+180))

#ifdef Ddec
  d11 H2_LOCK
  d11 LOCKH_OFF
; d11 LOCKDEC_OFF        /* use statement for earlier hardware */
#endif

HaltAcqu, 1m
exit

ph0=1
ph1=(6) 0 1 2 3 4 5
ph2=1 3
ph3={{0}*6}^2
ph4={{0}*12}^2
ph5=0 2
ph11={0 1 0 1 1 0 1 0}^2
ph12={0 3 0 3 3 0 3 0}^2
ph13=ph12+ph0
ph14=ph11-ph0
ph15={{{0 2}^2}^2}^2
ph25=0
ph26=0
ph27=1
ph28=2
ph29=3
ph31={{{0 2}*3}^2}^2

;pl1 : tpwr - power level for pwh
;pl2 : dhpwr - power level for 13C pulse pwc (p2)
;pl4  : power level for 2H high power pulses
;pl10 : tsatpwr - power level for presat
;pl11 : tpwrmess - power level for Messerle purge
;pl15 : power level for 1H CPMG pulses pwh_cp
;pl21 : dpwr - power level for  13C decoupling cpd2
;pl41 : power level for 2H waltz decoupling
;p10 : 1000usec water flip-back
;sp10 : water flip-back (on H2O)
;spw14 : power level for eburp1 pulse
;spnam14: eburp1 pulse on water
;p1 : pwh
;p2 : pwc
;p4 : 2H high power pulse
;p14 : eburp1 pulse width, typically 7000u
;p15 : 1H pw for CPMG pulses
;p50 : gradient pulse 50                                [1000 usec]
;p51 : gradient pulse 51                                [400 usec]
;p52 : gradient pulse 52                                [600 usec]
;p53 : gradient pulse 53                                [300 usec]
;p54 : gradient pulse 54                                [500 usec]
;p55 : gradient pulse 55                                [300 usec]
;p56 : gradient pulse 56                                [500 usec]
;p57 : gradient pulse 57                                [800 usec]
;pcpd2 : 13C pulse width for 13C decoupling
;pcpd4 : 2H pulse width for 2H decoupling
;d1 : Repetition delay D1
;d3 : taua ~1/(4*JCH)  ~1.8-2ms
;d4 : taub - set to 1/4JHC = 2.0 ms
;d6 : time_T2 CPMG duration <= 40ms
;d11 : delay for disk i/o, 30ms
;d16 : gradient recovery delay, 200us
;d19: delay for binomial water suppression, = 1/(4*|cnst1|)
;cpd2 : 13C decoupling during t2 according to program defined by cpdprg2
;cpd4 : 2H decoupling during t1
;cpdprg2 : 13C decoupling during t2
;cpdprg4 : 2H decoupling during t1
;cnst1 : offset of methyls from water (0.8 ppm - 4.7 ppm, in Hz)
;cnst4 : power level for 2H high power pulses
;cnst10: water frequency for presat
;cnst14 : power level in w for 1H selective pulse
;cnst15 : power in w for 1H CPMG pulses
;cnst21 : power in w for 13C dec
;cnst31 : minimum nuCPMG for given time_T2
;cnst41 : power level for 2H decoupling
;l1 : counter for the ncyc_cp values for cpmg
;l2 : actual value of ncyc_cp
;l3 : flag to switch between IP and AP for -Dcal_flg
;inf1 : 1/SW(X) = 2*DW(X)
;in0 : 1/(2*SW(x))=DW(X)
;nd0 : 2
;ns : 6*n
;FnMODE : States-TPPI, TPPI, States

;for z-only gradients:
;gpz0: 20%
;gpz1: 25%
;gpz2: 30%
;gpz3: -25%
;gpz4: 50%
;gpz5: -40%
;gpz6: -75%
;gpz7: -80%

;use gradient files:
;gpnam0: SMSQ10.32
;gpnam1: SMSQ10.32
;gpnam2: SMSQ10.32
;gpnam3: SMSQ10.32
;gpnam4: SMSQ10.32
;gpnam5: SMSQ10.32
;gpnam6: SMSQ10.32
;gpnam7: SMSQ10.32

;zgoptns: Dfsat Dmess_flg Dfscuba Dwater_flg Dwtg_flg Df1180 DDdec Dcomp_flg Dmq_flg Dcal_flg

error: pathspec 'QQ_CPMG.cw' did not match any file(s) known to git


adding untested hsqct2etf3gpsi3d.cw

adding untested hsqct2etf3gpsi3d.cw

;Clean-up gradient pairs added in SE block, bipolar gradients during t1
;Delays adjusted for zero first-order phase correction
;
;hsqct2etf3gpsi3d
;avance-version (07/04/04)
;3D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;for measuring N-15 T2 relaxation times
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition
;using f3 - channel
;using flip-back pulse
;as pseudo3D
;(use parameterset HSQCT2ETF3GPSI3D)
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p22=p21*2"
"d11=30m"
"d12=20u"
"d24=1s/(cnst4*cnst11)"
"d25=1s/(cnst4*cnst12)"
"d26=1s/(cnst4*4)"

"d21=450u"

"d31=(p30*16+d21*32)"

"d10=10u"

"in10=inf2/4"

"DELTA2=p16+d16+8u+d12+de-0.6366*p1"
"DELTA3=d21-p2/2"
"DELTA4=d25-p16-d16"
"DELTA5=d24-p16-d16-4u"
"DELTA6=d26-p16-d16-4u"

#   ifdef LABEL_CN
"DELTA1=d25-p16-d16-larger(p2,p8)-d10*2-p21*4/3.1415"
"spoff13=bf2*((cnst21+cnst22)/2000000)-o2"
#   else
"DELTA1=d25-p16-d16-p2-d10*2-p21*4/3.1415"
#   endif /*LABEL_CN*/

"spoff1=0"

aqseq 312

1 ze
  d11 pl16:f3 st0
2 6m do:f3
3 3m
4 d1

  (p1 ph1)
  d26 pl3:f3
  (center (p2 ph1) (p22 ph1):f3 )
  d26 UNBLKGRAD
  (p1 ph2)

  4u pl0:f1
  (p11:sp11 ph1:r):f1		; flipback(+x), +y -> +z
  4u
  p16:gp1
  d16 pl1:f1

  (p21 ph3):f3
  d25
  (center (p2 ph1) (p22 ph6):f3 )	; water -> -z
  d25 pl23:f3

6 d21
  (p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6):f3
  DELTA3
  (p2 ph1)				; water -> +z
  DELTA3
  (p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6):f3
  d21*2
  (p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6):f3
  DELTA3
  (p2 ph8)				; water -> -z
  DELTA3
  (p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6 d21*2 p30 ph6):f3
  d21
  lo to 6 times c

  p16:gp2*-1*EA
  d16
  DELTA4 pl3:f3
  (p22 ph6):f3

  2u
  d10 gron0
  d10 gron0*-1
  8u groff

#   ifdef LABEL_CN
  (center (p2 ph7) (p8:sp13 ph1):f2 )	; water -> +z
#   else
  (p2 ph7)
#   endif /*LABEL_CN*/

  2u
  d10 gron0
  d10 gron0*-1
  8u groff

  p16:gp2*EA
  d16
  DELTA1

  (center (p1 ph1) (p21 ph4):f3 )	; water -> -y
  4u
  p16:gp4
  d16
  DELTA5
  (center (p2 ph1) (p22 ph1):f3 )	; water -> +y
  DELTA5
  4u
  p16:gp4
  d16
  (center (p1 ph2) (p21 ph5):f3 )	; water -- +y
  4u
  p16:gp5
  d16
  DELTA6
  (center (p2 ph1) (p22 ph1):f3 )	; water -> -y
  DELTA6
  4u
  p16:gp5
  d16
  (p1 ph1)				; water -> -z
  DELTA2
  (p2 ph1)				; water -> +z
  4u
  p16:gp3
  d16
  d12 pl16:f3
  4u  BLKGRAD
  goscnp ph31 cpd3:f3

  3m do:f3
  3m st ivc
  lo to 3 times nbl

  3m ipp3 ipp4 ipp5 ipp6 ipp7 ipp31
  lo to 4 times ns

  d1 mc #0 to 4
     F1QF()
     F2EA(igrad EA & ip5*2 & rpp3 rpp4 rpp5 rpp6 rpp7 rpp31, id10 & ip3*2 & ip6*2 & ip31*2)
  d31
exit

ph0=0
ph1=0
ph2=1
ph3=0 2
ph4=0 0 2 2
ph5=3 3 1 1
ph6=0 0 0 0 2 2 2 2
ph7=0 0 2 2
ph8=2
ph31=0 2 2 0

;pl0 : 120dB
;pl1 : f1 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl16: f3 channel - power level for CPD/BB decoupling
;pl23: f3 channel - power level for TOCSY-spinlock
;sp1 : f1 channel - shaped pulse  90 degree
;sp13: f2 channel - shaped pulse 180 degree  (Ca and C=O, adiabatic)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p11: f1 channel -  90 degree shaped pulse
;p16: homospoil/gradient pulse
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p30: f3 channel - 180 degree pulse at pl23
;d1 : relaxation delay; 1-5 * T1
;d10 : incremented delay                             [3 usec]
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21: echo delay                                     [450 usec]
;d24: 1/(4J)YH for YH
;     1/(8J)YH for all multiplicities
;d25: 1/(4J)YH for YH
;     1/(8J)YH for all multiplicities
;d26: 1/(4J(YH))
;d31: length of single cpmg loop
;cnst4: = J(YH)
;cnst11: for multiplicity selection = 4 for NH, 8 for all multiplicities
;cnst12: for multiplicity selection = 4 for NH, 8 for all multiplicities
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;vc : variable loop-coounter for T2 delay, taken from vc-list
;inf2: 1/SW(X) = 2 * DW(X)
;in10: 1/(2 * SW(X)) = DW(X)
;nd10: 2
;NS: 2 * n
;DS: >= 16
;td1: number of delays in vc-list
;td2: number of experiments in F2
;NBL: = td1
;FnMODE: QF in F1
;FnMODE: echo-antiecho in F2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz0: 1-2%
;gpz1: 30%
;gpz2: 80%
;gpz3: 16.2%
;gpz4: 7%
;gpz5: -5%

;use gradient files:
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100

                                          ;preprocessor-flags-start
;LABEL_CN: for C-13 and N-15 labeled samples start experiment with
;             option -DLABEL_CN (eda: ZGOPTNS)
                                          ;preprocessor-flags-end

;$Id: hsqct2etf3gpsi3d,v 1.5 2007/04/11 13:34:30 ber Exp $

error: pathspec 'hsqct2etf3gpsi3d.cw' did not match any file(s) known to git


add hmqc_CH3_1Hcsa.cw

add hmqc_CH3_1Hcsa.cw

;IPAP HMQC for measurement of 1H CSA via 1H CSA/1H-13C DD CCR
; set td1 = 2*number of relaxation time points
;Chris Waudby, July 2020
;
;avance-version (07/04/04)
;HMQC
;2D H-1/X correlation via heteronuclear zero and double quantum
;   coherence
;phase sensitive
;with decoupling during acquisition
;
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d3=1s/(cnst2*8)"
"d11=30m"
"d12=20u"
"d13=4u"

"in0=inf2/2"
"d0=in0/2-0.63662*p3-2*p1"

; loop counter for IPAP
"l1=0"

define delay vdmin
"vdmin=2*(p2+4u+p17+d16)"

"acqt0=0"
baseopt_echo

aqseq 312

1 ze
  vdmin
  d11 ph1:f1 pl3:f2
2 d11

  20u
  "TAU1=vd*0.25-4u-p17-d16-p3"
  "TAU2=vd*0.25-p3-p1"
  "TAU3=vd*0.25-p1-4u-p17-d16-p3"
  "TAU4=vd*0.25-p3"

# ifdef OFFRES_PRESAT
    30u fq=cnst21(bf hz):f1
# endif /*OFFRES_PRESAT*/

  ; relaxation period
  d12 pl9:f1
  d1 cw:f1 ph29
  d13 do:f1
  d12 pl1:f1 pl2:f2
  30u fq=0:f1
  50u UNBLKGRAD

  (p3 ph1):f2    ; crush eq'm 13C magnetisation
  d13
  p16:gp1
  d16

  ; start main sequence
  (p1 ph1):f1  ; INEPT
  "DELTA1=d2-p16-d16+0.6366*p1"
  DELTA1
  p16:gp2
  d16

  ; purge element
  (p3 ph11):f2
  "DELTA=d3-p17-d16-larger(p1,p3)"
  DELTA
  p17:gp3
  d16
  (center (p2 ph1):f1 (p4 ph1):f2 )
  DELTA
  p17:gp3
  d16
  (p3 ph12):f2

  ; t1 evolution
  d0
  (p1 ph13):f1
  (p2 ph14):f1
  (p1 ph13):f1
  d0
  (p3 ph15):f2

  ; relaxation period
  "TAU=vd*0.5-p17-d16-p2-4u"
  TAU
  4u
  p17:gp4
  d16
  (p1 ph1):f1
  (p2 ph2):f1
  (p1 ph1):f1
  4u
  p17:gp4
  d16
  TAU

  ; IPAP back-transfer
  if "l1 % 2 == 0" {
  ; IP
  "DELTA2=d2*0.5-p16-d16-p3"
  p16:gp2
  d16
  DELTA2
  p4:f2 ph1
  "DELTA3=d2*0.5-p3-4u"
  DELTA3
  4u BLKGRAD
  } else {
  ; AP
  "DELTA2=d2-p16-d16-p3-4u"
  p16:gp2
  d16
  DELTA2
  p3:f2 ph1
  4u BLKGRAD
  }

  ; acquisition
  go=2 ph31
  d11 mc #0 to 2
    F1I(iu1, 2)
    F1QF(ivd)
    F2PH(ip15, id0)

  4u BLKGRAD
exit

ph1= 0
ph2= 1
ph11=0 2
ph12=1 1 1 1 3 3 3 3
ph13=0 0 1 1 2 2 3 3
ph14=1 1 2 2 3 3 0 0
ph15=0
ph29=0
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl9 : f1 channel - power level for presaturation
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse
;p17: gradient pulse [300 usec]
;p22 : f3 channel - 180 degree high power pulse
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)CH
;d3 : 1/(8J)CH
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(CH)
;cnst21: frequency in Hz for off-res presat
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 8 * n
;DS: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ

;for z-only gradients:
;gpz1: 31%
;gpz2: 7%
;gpz3: -40%
;gpz4: 29%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.32
;gpnam4: SMSQ10.32
                                          ;preprocessor-flags-start
;SINGLEDWELL: for initial sampling delay of one dwell-time with
;	    option -DSINGLEDWELL (eda: ZGOPTNS)
;OFFRES_PRESAT: for off-resonance presaturation, set cnst21=o1(water)
;	    option -DOFFRES_PRESAT (eda: ZGOPTNS)
                                          ;preprocessor-flags-end

;$Id: hmqcphpr,v 1.4 2007/04/11 13:34:30 ber Exp $

error: pathspec 'hmqc_CH3_1Hcsa.cw' did not match any file(s) known to git


adding noesynoesyfpgpphrs19.cw

adding noesynoesyfpgpphrs19.cw

;noesynoesyfpgpphrs19.cw
;avance-version (12/01/11)
;2D homonuclear correlation via dipolar coupling
;dipolar coupling may be due to noe or chemical exchange.
;phase sensitive
;water suppression using 3-9-19 pulse sequence with gradients
;using flip-back pulse
;with radiation damping suppression using gradients in t1
;
;M. Piotto, V. Saudek & V. Sklenar, J. Biomol. NMR 2,
;   661 - 666 (1992)
;V. Sklenar, M. Piotto, R. Leppik & V. Saudek, J. Magn. Reson. A102,
;   241 -245 (1993)
;G. Lippens, C. Dhalluin & J.-M. Wieruszeski, J. Biomol. NMR 5,
;   327-331 (1995)
;V. Sklenar, J. Magn. Reson. A114, 132-135 (1995)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"d12=20u"

"in0=inf1/2"
"in10=inf2/2"

"d0=in0/2-p1*2/3.1416-1u"
"d10=in10/2-p1*2/3.1416-1u"

"DELTA1=d8"
"DELTA2=d8-p16-d16-p11-d12-4u"

"TAU=p1*2/3.1416-(p0-p27)*0.231-de+46u"

aqseq 312
"acqt0=0"
baseopt_echo

1 ze
2 d1
3 d12 pl1:f1

  50u UNBLKGRAD
  ; t1 evolution
  p1 ph1
  d0 gron0
  d0 gron0*-1
  2u groff
  p1 ph2

  ; first mixing time
  DELTA1

  ; t2 evolution
  p1 ph11
  d10 gron0
  d10 gron0*-1
  2u groff
  p1 ph2

  ; second mixing time
  DELTA2

  ; final readout
  p16:gp1
  d16 pl0:f1
  (p11:sp1 ph3:r):f1
  4u
  d12 pl1:f1
  p1 ph4
  50u pl18:f1
  p16:gp2
  d16
  p27*0.231 ph5
  d19*2
  p27*0.692 ph5
  d19*2
  p27*1.462 ph5
  d19*2
  p27*1.462 ph6
  d19*2
  p27*0.692 ph6
  d19*2
  p0*0.231 ph6
  TAU
  p16:gp2
  d16
  4u BLKGRAD
  go=2 ph31
  d1 mc #0 to 2
       F1PH(calph(ph1, +90), caldel(d0, +in0))
       F2PH(calph(ph11, +90), caldel(d10, +in0))
exit

ph1= 0 2
ph11=0 0 2 2
ph2= 0
ph3= 2 2 2 2 0 0 0 0
ph4= 0 0 0 0 2 2 2 2
ph5= 0
ph6= 2
ph31=0 2 2 0 2 0 0 2

;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl18: f1 channel - power level for 3-9-19-pulse (watergate)
;sp1: f1 channel - shaped pulse  90 degree
;p0 : f1 channel -  90 degree pulse at pl18
;                      use for fine adjustment
;p1 : f1 channel -  90 degree high power pulse
;p11: f1 channel -  90 degree shaped pulse
;p16: homospoil/gradient pulse
;p27: f1 channel -  90 degree pulse at pl18
;d0 : incremented delay (2D, min >= 6usec)
;d1 : relaxation delay; 1-5 * T1
;d8 : mixing time
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d19: delay for binomial water suppression
;     d19 = (1/(2*d)), d = distance of next null (in Hz)
;inf1: 1/SW = 2 * DW
;in0: 1/(2 * SW) = DW
;nd0: 1
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ

;use gradient ratio:    gp 0 : gp 1 : gp 2
;                          2 :   50 :   30

;for z-only gradients:
;gpz0: 2%
;gpz1: 50%
;gpz2: 30%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

;$Id: noesyfpgpphrs19,v 1.12 2012/01/31 17:49:27 ber Exp $

error: pathspec 'noesynoesyfpgpphrs19.cw' did not match any file(s) known to git


adding stebphmqcgpphpr2.3d.cw

adding stebphmqcgpphpr2.3d.cw

; stebphmqcgpphpr2.3d.cw
; 13C methyl diffusion measurement
; pseudo-3D
; With water saturation during diffusion delay
; using baseopt
;
; With proton stimulated gradient-echo prior to HMQC
; Lit. Protein Engineering, Design & Selection, 24, 99-103 (2011)
;
; Removal of 13C equilibrium magnetisation (for methyl TROSY)
; Addition of clean-up gradient-pair
; Delays adjusted for zero first-order phase correction
; With options for 15N decoupling and 90,-180 or 180,-360 phase corr.
;
;hmqcphpr
;avance-version (07/04/04)
;HMQC
;2D H-1/X correlation via heteronuclear zero and double quantum
;   coherence
;phase sensitive
;with decoupling during acquisition
;
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple_d>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define list<gradient> diff=<Difframp>

"p2=p1*2"
"p22=p21*2"
"d2=1s/(cnst2*2)"
"d11=30m"
"d12=20u"
"d13=4u"

"in0=inf2/2"

"d0=in0/2-0.63662*p3-p1"

"DELTA1=d2-p16-d16"
"DELTA2=d2-p16-d16-d12-4u-0.6366*p1-de"
"DELTA3=d20-2*p30-p19-3*d16-p2-2*p1-d12-d13"

"acqt0=de"

aqseq 312

1 ze
2 d11 do:f2
  4u BLKGRAD
3 d12 pl9:f1
  d1 cw:f1 ph29
  d13 do:f1
  d12 pl1:f1 pl2:f2
  4u UNBLKGRAD
  (p3 ph1):f2
  d13
  p16:gp1
  d16
;----------- STE element (1) -------------
  p1 ph1
  p30:gp6*diff
  d16
  p2 ph5
  p30:gp6*-1*diff
  d16
  p1 ph1
  p19:gp4
  d16 pl9:f1
  DELTA3 cw:f1 ph29
  d13 do:f1

;----------- STE element (2) -------------
  d12 pl1:f1
  p1 ph2
  p30:gp6*diff
  d16
  p2 ph6
  p30:gp6*-1*diff
  d16
;--------------- HMQC ----------------

  DELTA1
  p16:gp2
  d16
  p3:f2 ph3
  d0
  (p2 ph1)
  d0
  p3:f2 ph4
  d12 pl12:f2
  p16:gp2
  d16
  4u BLKGRAD
  DELTA2
  go=2 ph31 cpd2:f2
  d11 do:f2 mc #0 to 2
	F1QF(igrad diff)
	F2PH(ip3 & ip29, id0)
  4u BLKGRAD
exit

ph1= 0
ph2= 2
ph3= 0 2
ph4= 0 0 2 2
ph5= 0 0 0 0 2 2 2 2
ph6= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph29=0
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl9 : f1 channel - power level for presaturation
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse
;p19: gradient pulse 2 (spoil gradient)
;p30: gradient pulse (little DELTA * 0.5)
;p22 : f3 channel - 180 degree high power pulse
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)CH
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;d20: diffusion time (big DELTA)
;cnst2: = J(CH)
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 16 * n
;DS: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;use gradient ratio:    gp 1  : gp 2  : gp6  : gp4
;                       -17.13: -13.17: var  :-15.37

;for z-only gradients:
;gpz1: -17.13%
;gpz2: -13.17%
;gpz4: -15.37%
;gpz6: 100%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam6: SMSQ10.100

                                          ;preprocessor-flags-start
;LABEL_CN: for C-13 and N-15 labeled samples start experiment with
;             option -DLABEL_CN (eda: ZGOPTNS)
;HALFDWELL: for initial sampling delay of half a dwell-time with
;	    option -DHALFDWELL (eda: ZGOPTNS)
                                          ;preprocessor-flags-end

;$Id: hmqcphpr,v 1.4 2007/04/11 13:34:30 ber Exp $

error: pathspec 'stebphmqcgpphpr2.3d.cw' did not match any file(s) known to git


cosyetgp_decoupled.cw added and tested

cosyetgp_decoupled.cw added and tested

;cosyetgp
; with continuous cpd in f2
; corrected for zero first order phase correction
; using baseopt
;avance-version (12/01/11)
;2D homonuclear shift correlation
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d11=30m"

"in0=inf1"

"d0=3u"

"DELTA=p16+d16+d0"
"DELTA1=p16+d16+8u"
"acqt0=0"

1 ze
  d11 pl12:f2
  d11 cpd2:f2
2 d11
3 d1
  50u UNBLKGRAD

  p1 ph1

  DELTA

  p2 ph2

  d0
  p16:gp1*EA
  d16

  p1 ph3

  DELTA1

  p2 ph2

  4u
  p16:gp2
  d16
  4u BLKGRAD

  go=2 ph31
  d11 mc #0 to 2 F1EA(calgrad(EA), caldel(d0, +in0) & calph(ph1, +180) & calph(ph31, +180))
  d11 do:f2
exit

ph1=0 2 2 0 1 3 3 1
ph2=0 2 0 2 1 3 1 3
ph3=0 2 0 2 1 3 1 3
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;p0 : f1 channel -  20 to 90 degree high power pulse
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                             [30 msec]
;d16: delay for homospoil/gradient recovery
;inf1: 1/SW = 2 * DW
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: echo-antiecho

;use gradient ratio:	gp 1 : gp 2
; 		          30 :   30

;for z-only gradients:
;gpz1: 30%
;gpz2: 30%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100

error: pathspec 'cosyetgp_decoupled.cw' did not match any file(s) known to git


added b_trosy_hzdqc.cw

added b_trosy_hzdqc.cw

;BEST-TROSY-H(Z/D)QC
;Chris Waudby, June 2018
;
;options:
; -DDQ = HDQC (otherwise runs HZQC)
; -DONE_D = first-row
; -DOFFRES_PRESAT = presat, pl9 on cnst21 (Hz bf)

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"d11=30m"
"d12=20u"
"d13=4u"
"d21=1s/(cnst4*2)"

"in0=inf1"
# ifdef ONE_D
"d0=2u"
#else
"d0=in0/2-p21*4/3.1415"
# endif /*ONE_D*/

"d2=p39-p39*cnst39-0.3633*p21"
"d3=0.5*p40-0.3633*p21"
"DELTA1=d21-p39*cnst39-p40*0.5-p16-d16"
"DELTA2=d21-0.3633*p21-p16-d16-0.5*p40"
"DELTA3=d21-p40-p16-d16"
"DELTA4=d21-0.5*p40-p16-d16-p21-de"
"acqt0=de"

# ifdef OFFRES_PRESAT
  "TAU=d1-d11-60u-d12*2-d13-d12-50u-p21-p16-d16"
# else
  "TAU=d1-d11-d12-50u-p21-p16-d16"
# endif /*OFFRES_PRESAT*/

"spoff23=bf1*(cnst19/1000000)-o1"
"spoff24=bf1*(cnst19/1000000)-o1"

"l0=1"  ; loop counter for shifting 1H 180 pulse between echo/anti-echoes

1 ze
  d11
2 d11

# ifdef OFFRES_PRESAT
  30u fq=cnst21(bf hz):f1
  d12 pl9:f1
  TAU cw:f1 ph29
  d13 do:f1
  d12 pl1:f1
  30u fq=0:f1
# else
  TAU
# endif /*OFFRES_PRESAT*/

3 d12 pl3:f3
  50u UNBLKGRAD

  ; purge Nz
  (p21 ph1):f3
  p16:gp0
  d16

  ; begin main sequence
  (lalign (p39:sp23 ph1) (d2 p21 ph11):f3 )
  DELTA1
  p16:gp1
  d16
  (center (p40:sp24 ph1) (p22 ph12):f3 )
  p16:gp1
  d16

  if "l0 %2 == 1"
     {
  (ralign (p40:sp24 ph1) (DELTA2 p21 ph13 d0 p21 ph1 d3):f3 )
  DELTA3
     }
  else
     {
  DELTA3
  (lalign (p40:sp24 ph1) (d3 p21 ph13 d0 p21 ph1 DELTA2):f3 )
     }

  p16:gp2
  d16
  (center (p40:sp24 ph1) (p22 ph1):f3 )
  p16:gp2
  d16
  DELTA4 BLKGRAD
  (p21 ph14):f3

  go=2 ph31
  d11 mc #0 to 2
     F1EA(iu0 & ip14*2 & ip31*2, id0)

exit

ph1=0
ph11=0 2 1 3
ph12=0 0 1 1
ph31=0 2 1 3
#ifdef DQ
ph13=1 3 2 0
ph14=3
#else /* ZQ */
ph13=3 1 0 2
ph14=1
#endif

;pl3 : f3 channel - power level for pulse (default)
;pl26: f3 channel - power level for CPD/BB decoupling (low power)
;sp23: f1 channel - shaped pulse 120 degree
;                   (Pc9_4_120.1000 or Q5.1000)
;sp24: f1 channel - shaped pulse 180 degree (Rsnob.1000)
;p16: homospoil/gradient pulse                       [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p39: f1 channel - 120 degree shaped pulse for excitation
;                      Pc9_4_120.1000 (120o)    (3.0ms at 600.13 MHz)
;                  (or Q5.1000 (90o)            (2.0ms at 600.13 MHz) )
;p40: f1 channel - 180 degree shaped pulse for refocussing
;                      Rsnob.1000               (1.0ms at 600.13 MHz)
;d0 : incremented delay (2D) = in0/2-p21*4/3.1415
;d1 : relaxation delay
;d11: delay for disk I/O                             [30 msec]
;d12: delay for power switching                      [20 usec]
;d16: delay for homospoil/gradient recovery
;d21 : 1/(2J)NH
;cnst4: = J(NH)
;cnst19: H(N) chemical shift (offset, in ppm)
;cnst39: compensation of chemical shift evolution during p39
;           Pc9_4_90.1000: 0.514
;           Pc9_4_120.1000: 0.529
;inf1: 1/SW(N) = 2 * DW(N)
;in0: 1/ SW(N) = 2 * DW(N)
;nd0: 1
;NS: 4 * n
;DS: 16
;td1: number of experiments
;FnMODE: Echo-AntiEcho

;use gradient ratio:	gp 0 : gp 1 : gp 2
;			-16 :  11 :    7

;for z-only gradients:
;gpz0: -16%
;gpz1: 11%
;gpz2:  7%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100

;Processing

;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1

error: pathspec 'b_trosy_hzdqc.cw' did not match any file(s) known to git


imaging_zg.cw

;imaging_zgesgp.cw
; 1D CSI
; td1 is hard-coded (look for size of r1d list)
;using baseopt
;avance-version (07/10/04)
;2D sequence
;water suppression using excitation sculpting with gradients
;T.-L. Hwang & A.J. Shaka, J. Magn. Reson.,
;   Series A 112 275-279 (1995)
;
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

; define list of gradient amplitudes, from -1 to 1, where final value is never reached
define list<grad_scalar, 384> r1d
;"td1=128"

"p2=p1*2"
"d12=20u"

"acqt0=de+4u+p19+d16+p1*0.6366"

1 ze
2 30m
#ifdef PRESAT
  d12 fq=cnst19(bf ppm):f1 ; off-resonance decoupling
  d12 pl9:f1 BLKGRAD
  d1 cw:f1
  4u do:f1
  4u pl1:f1
  d12 fq=0:f1 ; restore 1H frequency
#else
  d12 pl1:f1 BLKGRAD
  d1
  8u
#endif
  50u UNBLKGRAD
  p1 ph1

  ; imaging gradient
  p19:gp3*r1d
  d16
  4u BLKGRAD

  ; acquisition
  go=2 ph31
  30m mc #0 to 2 F1QF(r1d.inc)
  4u BLKGRAD
exit

ph1=0
ph31=0

;pl0 : 120dB
;pl1 : f1 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p19: imaging gradient [200 usec]
;d1 : relaxation delay; 1-5 * T1
;d12: delay for power switching                             [20 usec]
;d16: delay for homospoil/gradient recovery
;NS: 1 * n, total number of scans: NS * TD0
;DS: 4

;gpz3: 17%

;use gradient files:
;gpnam3: SINE.32

error: pathspec 'imaging_zg.cw' did not match any file(s) known to git


stebpgp1s19.kinetic.cw added

stebpgp1s19.kinetic.cw added

;pseudo-3D for kinetic measurements
; Delays in final spin-echo adjusted to give zero first-order phase corr.
; Use for flat baselines (with baseopt, Chris Waudby October 2016)
;
;stebpgp1s19
;avance-version (07/05/08)
;2D sequence for diffusion measurement using stimulated echo
;using bipolar gradient pulses for diffusion
;using 1 spoil gradient
;water suppression using 3-9-19 pulse sequence with gradients
;with added presat water suppression
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple_d>

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define list<gradient> diff=<Difframp>

"p2=p1*2"

"DELTA1=d20-p1*2-p2-p30*2-d16*3-p19"

"TAU=0.6366*p1+8u"

"acqt0=0"
aqseq 321

1 ze
2 d11
  4u pl9:f1
  d1 cw:f1 ph1
  4u do:f1

  50u pl1:f1 UNBLKGRAD
  p1 ph1
  p30:gp6*diff
  d16
  p2 ph2
  p30:gp6*-1*diff
  d16
  p1 ph3
  p19:gp7
  d16
  DELTA1
  p1 ph4
  p30:gp6*diff
  d16
  p2 ph2
  p30:gp6*-1*diff
  d16
  p16:gp1
  d16 pl18:f1
  TAU
  p27*0.231 ph5
  d19*2
  p27*0.692 ph5
  d19*2
  p27*1.462 ph5
  d19*2
  p27*1.462 ph6
  d19*2
  p27*0.692 ph6
  d19*2
  p0*0.231 ph6
  4u
  p16:gp1
  d16
  4u BLKGRAD
  go=2 ph31
  d1 mc #0 to 2
    F2QF(igrad diff)
    F1QF()
exit

ph1= 0
ph2= 0 0 0 0  2 2 2 2
ph3= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph4= 0 1 2 3
ph5= 0
ph6= 2
ph31=0 1 2 3  0 1 2 3  2 3 0 1  2 3 0 1

;pl1: f1 channel - power level for pulse (default)
;pl9: f1 channel - power level for presaturation
;pl18: f1 channel - power level for 3-9-19-pulse (watergate)
;p0 : f1 channel -  90 degree pulse at pl18
;                      use for fine adjustment
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p16: gradient pulse (WATERGATE)
;p19: gradient pulse 2 (spoil gradient)
;p27: f1 channel -  90 degree pulse at pl18
;p30: gradient pulse (little DELTA * 0.5)
;d1 : relaxation delay; 1-5 * T1
;d16: delay for gradient recovery
;d19: delay for binomial water suppression
;     d19 = (1/(2*d)), d = distance of next null (in Hz)
;d20: diffusion time (big DELTA)
;NS : 8 * n
;DS : 4 * m
;td1: number of experiments
;FnMODE: QF
;        use xf2 and DOSY processing

;use gradient ratio:    gp 1 : gp 6 : gp7
;                        -20 :  100 : -17.13

;for z-only gradients:
;gpz1: -20%
;gpz6: 100%
;gpz7: -17.13% (spoil)

;use gradient files:
;gpnam1: SINE.100
;gpnam6: SINE.100
;gpnam7: SINE.100

;use AU-program dosy to calculate gradient-file Difframp

;$Id: stebpgp1s19,v 1.4.6.1 2007/05/09 09:36:59 ber Exp $

error: pathspec 'stebpgp1s19.kinetic.cw' did not match any file(s) known to git


sfhmqcnoesysfhmqcgpph.4d.cw added

sfhmqcnoesysfhmqcgpph.4d.cw added

;4D HCCH SFHMQC-NOESY-SFHMQC
;for methyl-methyl NOES
;Option for NUS using Topspin 3
;Derived from hmqcnoesyhmqcccgpphpr.jk
;Chris W, July 2018

;F1(H) -> F2(C[mq],t1) ---NOE--> F1(H) -> F2(C[mq],t2) -> F1(H,t3)
;
;Indirect evolution order is t2, t1 (13Cdir, 13Cnoe)
;Uses half-dwell first-point delay by default in all indirect dims
;Option for off-res presat
;Options for 2D planes in each 13C dim
;  (set both to get 1D or 2D HH plane with no 13C phase evolution)
;Removal of 13C equilibrium magnetisation
;Delays adjusted for zero first-order phase correction in acqusition dim

;$CLASS=HighRes
;$DIM=4D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d11=45m" ; for disk access

;------------indirect 1H dim (F1)
"in0=inf1/2"
"d0=in0/2-p3"

;------------options for first (in transfer pathway) 13C dim (F2)
"in10=inf2/2"		; first 13C dim (NB two d10 delays present)
"d10=in10/2-p3*4/3.1415"

;------------options for third (in transfer pathway) 13C dim (F3)
"in30=inf3"		; second 13C dim (NB only one d30 delay present)
"d30=in30/2-p3*4/3.1415"

"spoff23=bf1*(cnst19/1000000)-o1"
"spoff24=bf1*(cnst19/1000000)-o1"
"spoff25=bf1*(cnst19/1000000)-o1"

"TAU=d8-p16*2-d16*2-p3-8u"  ; noe mixing time

; delays for first SFHMQC
"d5=d2-p39*cnst39-4u-p16-d16"
"DELTA1=d5+p3+p4-p40*0.5"

; delays for second SFHMQC
"DELTA2=d2-p16-d16-p39*cnst39"
"DELTA3=p39*cnst39-de-4u"
"acqt0=de"

;aqseq 4321	; for info only

1 ze
  d11 pl12:f2
  4u BLKGRAD
2 d11 do:f2
  d1
  50u UNBLKGRAD

;-------------------------kill equm 13C magnetisation

  (p3 ph1):f2
  4u
  p16:gp1
  d16*2

;-------------------------start first 13C HMQC element

  (p41:sp25 ph11):f1
  4u
  p16:gp2
  d16

  ; 1H F1 and 13C F2 evolution (MQ)
  (lalign
    (DELTA1 2*d0 d10 p40:sp24 ph2):f1
    (d5 p3 ph12 d0 p4 ph1 d0 2*d10 p3 ph1 d5):f2
  )

  4u
  p16:gp2
  d16
  (p41:sp25 ph1):f1

;------------------------start NOE period
  4u
  p16:gp3*0.71
  d16
  (p3 ph1):f2
  4u
  p16:gp3
  d16

  TAU

;------------------------start second 13C HMQC element
  (p39:sp23 ph1):f1
  p16:gp4
  d16

  (center (p40:sp24 ph2):f1 (DELTA2 p3 ph13 d30 p3 ph1 DELTA2):f2 )

  DELTA3
  p16:gp4
  d16 pl12:f2
  4u BLKGRAD
  go=2 ph31 cpd2:f2

#ifndef NUWS
  d11 do:f2 mc #0 to 2
    F3PH(ip13, id30)
	F2PH(rp13 & rd30 & ip12, id10)
	F1PH(rp13 & rd30 & rp12 & rd10 & ip11, id0)
#else
  d11 do:f2 mc #0 to 2
	F1PH(calph(ph11), caldel(d0))
	F2PH(calph(ph12), caldel(d10))
	F3PH(calph(ph13), caldel(d30))
#endif /*NUS*/
  4u BLKGRAD

exit

ph1= 0
ph2= 1
ph11=0
ph12=0 2
ph13=0 0 2 2
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl9 : f1 channel - power level for presaturation
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse                       [1 msec]
;p19: second gradient pulse                          [250 usec]
;p22: f3 channel - 180 degree high power pulse
;p28: f1 channel - trim pulse                        [1 msec]
;d0 : incremented delay (4D)
;d10: incremented delay (4D)
;d20: decremented delay (4D)
;d28: incremented delay (4D)
;d30: incremented delay (4D)
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)CH
;d8 : mixing time
;d11: delay for disk I/O                             [45 msec]
;d12: delay for power switching                      [20 usec]
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(CH)
;cnst21: frequency in Hz for off-res presat
;inf1: 1/SW(H)
;inf2: 1/SW(C) = 2 * DW(C)
;inf3: 1/SW(C) = 2 * DW(C)
;in0: 1/(2 * SW(C)) = DW(C)
;in10: 1/(2 * SW(C)) = DW(C)
;in30: 1/(2 * SW(H)) = DW(H)
;nd0: 2
;nd10: 2
;NS: 4 * n
;DS: 16
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or States) in F1
;FnMODE: States-TPPI (or States) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 31%
;gpz2: 29%
;gpz3: 23%
;gpz4: 13%
;gpz5: 43%
;gpz6: 19%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SINE.50
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.50
;gpnam6: SINE.100

                                          ;preprocessor-flags-start
;NUS: for non-uniform sampling (Topspin 3)
                                          ;preprocessor-flags-end

error: pathspec 'sfhmqcnoesysfhmqcgpph.4d.cw' did not match any file(s) known to git